

ATARI
MUSEUM.NL

BASIC-PROGRAMMA'S VOOR DE ATARI 600XL/800XL

Peter Goode



Kluwer Software-reeks



**BASIC-programma's
voor de
Atari 600XL/800XL**

Peter Goode



BASIC-programma's voor de Atari 600XL/800XL



**Kluwer Technische Boeken B.V.
Deventer - Antwerpen**



Vertaald door: Marc Koetsier

Omslag: W. Niessink

**ISBN 90 201 1836 6
D/1985/0108/189**

**Oorspronkelijke titel: The ATARI 600XL Program Book
Uitgegeven bij: Phoenix Publishing Associates, Bushey, Herts.**

**© 1983 Peter Goode
© 1985 van de Nederlandse vertaling bij
Kluwer Technische Boeken B.V. - Deventer**

1e druk 1985

Niets uit deze uitgave mag worden verveelvoudigd en/of openbaar gemaakt door middel van druk, fotokopie, microfilm of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van de uitgever.

No part of this book may be reproduced in any form, by print, photoprint, microfilm or any other means without written permission from the publisher.

Ondanks alle aan de samenstelling van de tekst bestede zorg, kan noch de redactie noch de uitgever aansprakelijkheid aanvaarden voor eventuele schade, die zou kunnen voortvloeien uit enige fout, die in deze uitgave zou kunnen voorkomen.

Woord vooraf

Als u reeds in het bezit bent van een ATARI XL computer, of van plan bent er een aan te schaffen, heeft u waarschijnlijk reeds ontdekt dat deze computer meer potentieel biedt voor spelen, programmeren enz. in kleur en geluid dan andere machines.

Omdat wij op de hoogte zijn van de mogelijkheden van een ATARI XL, hebben wij een brede selectie programma's geschreven die gebruik maken van de faciliteiten die de computer biedt.

Als u van spelen houdt, dan vindt u voldoende gelegenheid om de kleuren en graphics te gebruiken terwijl u uw reacties en uw vingervlugheid uittest. Mocht u geïnteresseerd zijn in avontuurspelen dan hebben we een programma voor u dat een variatie is van Dungeons en Dragons. Wij hebben, in verband met ruimtegebrek, geen mogelijkheid gezien het programma volledig uitgebreid te brengen, maar door toepassing van meer geheugen of een andere opslagmogelijkheid kunt u een breder scala van omstandigheden gebruiken voor uw avontuurspel.

Om gebruikers te helpen de programma's te begrijpen, hebben we volledige 'hoe en waarom'-instructies toegevoegd aan de programma's, en voor degene die klaarstaat om een stap naar grotere dingen te wagen, programmeertips, waarvan wij denken dat ze wellicht interessant zijn.

Inhoud

Staafterieken	10
Schoonvegen.	13
Doolhofkruiper	15
Eenentwintigen.	19
Tovenaars en heksen.	23
Woordpuzzel	28
S.I.M.P.U.R.	31
Catalogus	36
Ruimtestad.	39
Dobbelen.	42
Missile Command	46
Bioritme	51
Muurstoter	54
Onderzeebootjager	57
Brandkastkraker	61
Fruitmachine	64
Flitsende graphics	69
Akkoordenorgel	72
Robot-eiland	74
Wereldvlaggen	78
Kruisraket-aanval.	82
Slagbal.	85
Eivanger	88
Zwarte gaten	90
Meteorietenstorm	94
Wegpiraat	97
Muurbreker.	99
Oceaanslag.	102
Wegversperring	106
Mijnenveld.	108
Schiettent	111
Anagrammen.	113
Eeuwkalender	116
Surround.	119
Hark the Herald Angels Sing	122



Enige opmerkingen vooraf

Als u voor het eerst met computers in aanraking komt en niet gewend bent programma-listings in te typen, onthoud dan dat het van groot belang is om op elk detail te letten. Er zijn geen snelle methodes!

Uw computer is zeer gevoelig voor zowel taal- als leestekenfouten en zal de programma's niet laten lopen als zelfs ook maar één detail ontbreekt.

Besteed veel aandacht aan het verschil tussen 0 (het cijfer nul) en de O (de letter O). De nul is duidelijk smaller in alle listings in dit boek, de nul op het scherm is doorstreept (Ø).

Alle programma's die in dit boek zijn opgenomen, zijn afgedrukt rechtstreeks uit een ATARI XL computer, dus er zitten geen programmafouten in.

Mocht u deze programma's willen bewaren op een cassette om te voorkomen dat u ze telkens weer moet intoetsen, verzeker u er dan van dat het programma werkelijk op de band staat voordat u verder gaat.

Altijd eerst opnemen, voor het spelen van het spel.

U zult zien dat geduld wordt beloond in deze wereld van de hoge technologie.

Een noot van de vertaler

Zoals gezegd zijn de programma-listings rechtstreeks uit een computer genomen waarin het spel, of programma dat de listing vertegenwoordigt op dat moment werkt... dit is ook zo. Alle programma's zijn getest op goede werking... en ze doen het allemaal. Mocht u, bij het intoetsen van deze listings moeilijkheden ondervinden, moeilijkheden die zich uiten als een niet werkend programma, VERGELIJK dan uw programma met dat wat in het boek staat.

Bij het vertalen van dit boek zijn alle programma's ingetoetst en overnieuw uitgeprint. Enkele programma's zullen u weinig moeilijkheden geven, andere daarentegen zullen u ongetwijfeld zo lang bezighouden met foutzoeken dat u zich zult afvragen of het wel de moeite waard is. Ik heb wél doorgezet en alles op het scherm gehad wat ik op het scherm wou hebben.

Ik heb wel enkele tips voor u:

We hebben al opgemerkt dat de letter 'O' en het getal NUL met elkaar verward kunnen worden. Let goed op het verschil! Bij vele programma's worden Player/Missiles gebruikt en de technische kant hiervan kent verschillende variabelen, zoals H1, H2 enz. Let op dat er niet H1 wordt ingetoetst. Dit komt neer op hetzelfde verwarrende verschijnsel tussen 1 en l, als tussen 'O' en NUL.

Bij gebruik van DATA-getallen in een machinetaalroutine is het eveneens oppassen geblazen. Alle getallen dienen net zo te worden ingetoetst als ze in het boek staan.

Vanwege druktechnische redenen zijn de listings afgedrukt in 53 lettertekens breedte. Let vooral op het juiste intypen van het aantal spaties in de diverse PRINT-opdrachten. Refereer steeds naar de letters in andere regels bij het bepalen van de juiste spatiëring van de PRINT-instructies.


Veel plezier!

Marc Koetsier

Staafigrafieken



Dit is een handig programma voor eenieder die vergelijkingen wil maken van bijvoorbeeld verkoopcijfers, temperaturen, koersschommelingen op de wereldmarkt of zomaar om te zien hoeveel men per maand uitgeeft aan Space Invaders of Pac-Man.



Hoe te gebruiken

Dit programma laat de gebruiker tot 80 gegevens grafisch weergeven en zal de gegevens in drie kleuren weergeven om vergelijkingen gemakkelijker te kunnen maken.

U krijgt een 'keuzemenu' te zien en een vraag hoeveel gegevens u wilt vergelijken. Toets het getal in en de computer vraagt u naar de gegevens of de waarde van elke 'staaf'.

Gegevens worden eenvoudig met de cijfertoetsen ingetypt. U hoeft geen andere toetsen te gebruiken.

Zowel de horizontale als de verticale assen worden volautomatisch door de computer opgebouwd bij het analyseren van het programma.

Om overnieuw te beginnen, op de spatiebalk drukken.

Programma

```

10 REM **** STAAFGRAFIEKEN ****
20 DIM H(80):N=0
30 C1=1:C2=3:C3=5
40 REM HOOFDMENU
50 GRAPHICS 1+16
60 POSITION 5,2:PRINT #6;"GRAFIEKEN"
70 POSITION 5,4:PRINT #6;"=====
80 POSITION 0,10:PRINT #6;"1) DATA TOEVOEGEN"
90 POSITION 0,12:PRINT #6;"2) TOON GRAFIEKEN"
100 POSITION 0,14:PRINT #6;"3) KLEUR VERANDEREN"
110 POSITION 0,16:PRINT #6;"4) EINDE PROGRAMMA"
120 OPEN #1,4,0,"K:"
130 GET #1,D:IF D<ASC("1") OR D>ASC("4") THEN 130
140 CLOSE #1:IF D=ASC("4") THEN GRAPHICS 0:END
150 ON D-48 GOSUB 170,310,570
160 GOTO 40
170 REM DATA TOEVOEGEN
180 TRAP 180:GRAPHICS 0:POKE 710,146
190 POKE 712,146:PRINT "HOEVEEL STAVEN ";
200 INPUT N:IF N<2 OR N>80 THEN 190
210 POKE 752,1:REM CURSOR WEG
220 FOR L=1 TO N
230 TRAP 230
240 POSITION 2,10:PRINT "STAAF: ";L;"="
      ":POSITION 11,10
250 INPUT D:POSITION 20,20
260 IF D<0 THEN 240
270 POKE 752,1:REM CURSOR UIT
280 H(L)=D:NEXT L
290 REM FOR Q=L TO 80:H(Q)=0:NEXT Q
300 RETURN
310 REM TOON GRAFIEKEN
320 TRAP 320
330 GRAPHICS 1+16:IF N=0 THEN PRINT #6;" ER IS GEEN
DATA":GOTO 490
340 GRAPHICS 7+16:SETCOLOR 0,C1,10:SETCOLOR 1,C2,10:S
ETCOLOR 2,C3,10
350 S=INT(160/N)
360 M=0
370 FOR L=1 TO N
380 IF H(L)>M THEN M=H(L)
390 NEXT L
400 U=95/M
410 C=0:L=1:FOR I=0 TO (S)*N-1 STEP S
420 C=C+1:IF C=4 THEN C=1
430 COLOR C
440 FOR J=0 TO S-1

```

```

450 PLOT J+I,95:DRAWTO J+I,(95-U*H(L))
460 NEXT J
470 L=L+1
480 NEXT I
490 GOSUB 510
500 RETURN
510 REM OP SPATIE WACHTEN
520 OPEN #1,4,0,"K:"
530 GET #1,D
540 IF D<>32 THEN 530
550 CLOSE #1
560 RETURN
570 REM KLEUREN SCHEMA
580 TRAP 580
590 GRAPHICS 0:POKE 710,4:POKE 712,4:POKE 709,0
600 POSITION 16,2:? "(1-15)":POSITION 2,5:PRINT "KLEU
R EEN ";
610 INPUT C1:IF C1<1 OR C1>15 THEN 600
620 POSITION 2,10:PRINT "KLEUR TWEE ";
630 INPUT C2:IF C2<1 OR C2>15 THEN 620
640 POSITION 2,15:PRINT "KLEUR DRIE ";
650 INPUT C3:IF C3<1 OR C3>15 THEN 640
660 RETURN

```

Schoonvegen




Misschien herinnert u zich nog dat de Amerikaanse Space Shuttle een 'vliegende baksteen' genoemd werd door de piloten, terwijl hij neerkwam richting aarde met 8000 km/uur zonder motor.

Wij hebben u hetzelfde probleem gegeven, met een toegevoegde moeilijkheidsfactor, door u neer te laten komen in een verlaten stad.

Wel hebben wij u een voordeeltje meegegeven.

U heeft een lading bommen aan boord waarmee u een baan kunt schoonvegen voor uw 'vliegende baksteen'.



Hoe te spelen

Linksboven op het beeldscherm gekomen dient u over de stad te vliegen en de wolkenkrabbers te verpletteren door op de spatiebalk te drukken en uw bommen af te werpen.

U kunt een moeilijkheidsgraad bij het begin van het spel, tussen de 1 en 10, instellen.

U kunt slechts een tweede of een derde bom laten vallen als de eerste iets heeft geraakt, dus het heeft geen zin zomaar op de spatiebalk te drukken.

Programmeertips

Om de hoogte van de gebouwen te veranderen, verandert u het getal na de '*' in regel 90.

Onthoud wel dat als u de hoogte vergroot, u uw responstijd verkort.

Programma

```

10 REM SCHOONVEGEN
20 GRAPHICS 0:POKE 710,0:PRINT "SNELHEID(1-10) ":PRIN
T "1 IS SNEL":PRINT "10 IS LANGZAAM":POSITION 16,0
30 INPUT SP:IF SP<1 OR SP>10 THEN 20
40 SC=0
50 REM STAD TEKENEN
60 GRAPHICS 3+16
70 COLOR 3
80 FOR X=0 TO 39
90 FOR Y=23 TO 23-INT(RND(1)*5+1) STEP -1
100 PLOT X,Y
110 NEXT Y
120 NEXT X
130 X=0:Y=0
140 E=0:F=0
150 A=X+1:B=Y
160 FOR L=1 TO 4*SP:NEXT L
170 IF A=40 THEN A=0:B=B+1
180 IF B=23 AND A=20 THEN 320
190 LOCATE A,B,C
200 IF C=3 THEN 380
210 COLOR 4:PLOT X,Y
220 COLOR 1:PLOT A,B
230 X=A:Y=B
240 IF PEEK(764)=33 AND F=0 AND Y<>23 THEN F=Y+1:E=X:
POKE 764,0:COLOR 3:PLOT E,F
250 IF F=0 THEN 150
260 G=F+1:IF G=24 THEN COLOR 4:PLOT E,F:F=0:GOTO 150
270 COLOR 4:PLOT E,F
280 LOCATE E,G,D:IF D=3 THEN SC=SC+10
290 COLOR 2:PLOT E,G
300 F=G
310 GOTO 150
320 REM GELAND
330 GRAPHICS 1+16:POSITION 3,10
340 PRINT #6;"U BENT GELAND"
350 POSITION 5,14:PRINT #6;"SCORE : ";SC
360 FOR L=1 TO 600:NEXT L
370 RUN
380 REM NEERGESTORT
390 FOR L=0 TO 250 STEP 3
400 SOUND 0,L,3,14:NEXT L:SOUND 0,0,0,0
410 GRAPHICS 1+16
420 POSITION 2,10
430 PRINT #6;"U SCORDE :";SC
440 POSITION 0,17:PRINT #6;"          <RETURN>"
450 OPEN #1,4,0,"K:"
460 GET #1,D:IF D<>155 THEN 460
470 CLOSE #1:RUN


```

Doolhofkruiper

Dit kleine monster zal u gauw laten weten of uw reacties en vingervlugheid zijn zoals u denkt.

U heeft de keuze uit twee doolhoven en u moet uzelf van links onder naar rechts boven zien te bewegen.

Zodra u even een muur raakt, is het met u gedaan.



Hoe te spelen

Gebruik de toetsen Z, X, / en + om naar boven, beneden, rechts en links te bewegen. U heeft de keuze uit twee doolhoven in het begin van het spel, en u kunt op toets 1 of 2 drukken om uw keuze kenbaar te maken.

Aan het eind van elke poging zal worden gevraagd of u nogmaals wilt kiezen of overnieuw wilt beginnen.

Als u de uitgang haalt krijgt u de score te zien.

Trouwens, wij denken dat nr. 1 de moeilijkste is van de twee. Kijk of u het met ons eens bent.

Programmeertips

Om volledig gebruik te maken van het geheugen van de computer hebben we een korte machinetaal-routine van regel 970 tot 1020, pas dus op bij het intoetsen van deze regels.

Programma

```

10 REM DOOLHOFKRUIPER
20 REM MET MACHINE/CODE BEWEGING
30 POKE 53248,0:POKE 53249,0
40 GOSUB 490
50 GOSUB 610
60 X=52:Y=175:GOSUB 760
70 POKE H2,180
80 M=0:POKE H1,X
90 CX=3
100 GOSUB 810
110 SC=5000
120 K=FEEK(764)
130 SC=SC-5
140 POKE 53278,1
150 IF K=23 OR K=2Z OR K=38 OR K=6 THEN M=K
160 IF M=22 THEN X=X+CX:POKE H1,X
170 IF M=23 THEN X=X-CX:POKE H1,X
180 IF M=38 THEN Z=USR(12288)
190 IF M=6 THEN Z=USR(12301)
200 IF PEEK(53252)<>0 THEN 240
210 IF PEEK(53260)<>0 THEN 310
220 FOR T=1 TO 20:NEXT T
230 GOTO 120
240 GRAPHICS 1+16:POSITION 2,8
250 POKE H1,0:POKE H2,0
260 PRINT #6;"HET IS NIET GELUKT"
270 POSITION 0,10
280 PRINT #6;"          <SPATIE>      "
290 IF PEEK(764)<>33 THEN 290
300 RUN
310 GRAPHICS 1+16:POSITION 3,10
320 POKE H1,0:POKE H2,0
330 PRINT #6;"een goede score"
340 POSITION 6,13:PRINT #6;"VAN ";STR$(SC)
350 POSITION 0,17
360 PRINT #6;"          <SPATIE>"
370 IF PEEK(764)<>33 THEN 370
380 RUN
390 REM DOOLHOF KIEZEN
400 GRAPHICS 1+16:POSITION 0,10
410 PRINT #6;"          DOOLHOF"
420 POSITION 0,16
430 PRINT #6;"          (1 OF 2)  "
440 OPEN #1,4,0,"K:"
450 GET #1,A:IF A<49 OR A>50 THEN 450
460 RESTORE 860:IF A=50 THEN RESTORE 1030
470 CLOSE #1
480 RETURN

```


```
490 REM DOOLHOF TEKENEN
500 GOSUB 390:GRAPHICS 5+16
510 SETCOLOR 2,8,4:COLOR 3
520 FOR Y=0 TO 40 STEP 8
530 FOR X=0 TO 64 STEP 8
540 READ D:IF D=1 THEN PLOT X,Y:DRAWTO X+8,Y
550 NEXT X:NEXT Y
560 FOR Y=0 TO 32 STEP 8
570 FOR X=0 TO 72 STEP 8
580 READ D:IF D=1 THEN PLOT X,Y:DRAWTO X,Y+8
590 NEXT X:NEXT Y
600 RETURN
610 REM PLAYER MISSILE GRAPHICS
620 PMBASE=40:POKE 54279,PMBASE
630 C1=704:H1=53248
640 POKE 559,62:POKE 53277,3
650 P1=PMBASE*256+1024
660 P2=P1+256:H2=53249:C2=705
670 POKE C1,88:POKE H1,0
680 POKE C2,47:POKE H2,0
690 FOR I=0 TO 255
700 POKE P1+I,0:POKE P2+I,0:NEXT I
710 POKE P2+42,129:POKE P2+49,129
720 POKE P2+43,66:POKE P2+48,66
730 POKE P2+44,36:POKE P2+47,36
740 POKE P2+45,24:POKE P2+46,24
750 RETURN
760 Q=Y+P1
770 POKE Q,24:POKE Q+1,24:POKE Q+2,24
780 POKE Q+3,255:POKE Q+4,255:POKE Q+5,24
790 POKE Q+6,24:POKE Q+7,24
800 RETURN
810 REM MACHINE/CODE INLEZEN
820 RESTORE 970
830 FOR L=12288 TO 12313
840 READ D:POKE L,D
850 NEXT L:RETURN
860 DATA 1,1,1,1,1,1,1,1,1
870 DATA 0,1,0,0,0,0,0,0,1
880 DATA 1,0,0,1,0,1,0,1,0
890 DATA 0,1,1,0,0,0,0,0,1
900 DATA 1,0,0,1,1,0,0,0,0
910 DATA 1,1,1,1,1,1,1,1,1
920 DATA 1,0,0,1,0,0,0,1,0,1
930 DATA 1,0,1,1,0,1,1,1,0,1
940 DATA 1,0,1,0,1,0,0,1,0,1
950 DATA 1,0,1,0,0,1,1,1,0,1
960 DATA 1,0,1,0,0,0,1,0,0,1
970 DATA 162,252,189,0,44
980 DATA 157,3,44,202,208,247
```

990 DATA 104,96
1000 DATA 162,0,189,3,44
1010 DATA 157,0,44,232,208,247
1020 DATA 104,96
1030 DATA 1,1,1,1,1,1,1,1,1
1040 DATA 0,1,1,1,0,0,1,0,0
1050 DATA 0,0,1,1,1,0,0,1,0
1060 DATA 0,1,1,1,0,0,0,0,0
1070 DATA 0,0,1,1,1,0,0,0,0
1080 DATA 1,1,1,1,1,1,1,1,1
1090 DATA 1,0,0,0,0,1,0,0,1,1
1100 DATA 1,1,0,0,0,1,1,0,1,1
1110 DATA 1,1,0,0,0,1,1,0,1,1
1120 DATA 1,1,0,0,0,1,1,1,1,1
1130 DATA 1,1,0,0,0,0,1,0,0,1

Eenentwintigen

Het is gokken geblazen als u het opneemt met geld en geluk tegen de ATARI croupier, het is het oude spel. Trouwens, u kunt niet vals spelen bij het laten zien van uw kaarten omdat de computer weet wat u heeft.

De bedoeling van het spel is, voor het geval u het nog niet weet, met zo weinig mogelijk kaarten een score van 21 te behalen. Als u te veel kaarten koopt en meer dan 21 haalt, bent u 'stuk' en verliest u uw inzet. Een aas telt voor 1 of 11 behalve als u twee azen ontvangt bij het geven.



Hoe te spelen

De computer geeft u uw eerste twee kaarten en vraagt wat u wilt:

'N' (Nog een kaart zonder de inzet te verhogen)

'H' (Houd uw kaarten of eventuele kaartentotaal)

'K' (Koop nog een kaart)

'O' (Opgeven als u denkt dat uw kaarten niet te gebruiken zijn.

Onthoud dat u uw inzet bent verloren en u nieuwe kaarten krijgt.)

Als u zoveel kaarten koopt dat u op 21 uitkomt, dan verslaat u de computer tenzij deze ook 21 heeft.

Druk de toets in die u wilt hebben voor uw keuze na elke inzet. Het is niet nodig om op RETURN te drukken.

Als u een kaart koopt, typ dan uw inzet in waarvoor u wilt wedden en druk daarna op RETURN.

Denk erom: als u verliest, hoeft u uw T-shirt niet in te leveren.

Programma

```

10 REM **** EEN EN TWINTIGEN ****
20 GRAPHICS 0
30 DIM C(52)
40 DIM P(10),M(2)
50 GOSUB 770
60 M=100
70 DIM K$(40)
80 OPEN #1,4,0,"K:"
90 DIM A$(1)
100 REM HOOFD ROUTINE
110 GRAPHICS 0
120 M=M-2:IF M<0 THEN 710
130 PRINT "U HEEFT ";M;" GULDEN"
140 BET=2
150 PRINT "UW KAARTEN :"
160 GOSUB 930
170 GOSUB 1010:PRINT " EEN ";K$
180 P(1)=V:GOSUB 1010:GOSUB 930
190 PRINT " EEN ";K$:P(2)=V:GOSUB 930
200 N=2
210 IF P(1)=1 THEN P(1)=11
220 IF P(2)=1 AND P(1)<>11 THEN P(2)=11
230 IF P(1)+P(2)=21 THEN PRINT :PRINT "          EEN EN
TWINTIG!!          "
240 SP=P(1)+P(2)
250 REM KEUZE SPELER
260 PRINT "(";SP;)"":PRINT "NEEM/HOUD/KOOP/PAS?"
270 GET #1,A:A$=CHR$(A)
280 IF A$<>"N" AND A$<>"H" AND A$<>"K" AND A$<>"P" TH
EN 270
290 IF A$="P" THEN PRINT "O.K U VERLIEST UW INZET":GO
SUB 930:GOTO 110
300 IF A$="H" THEN 410
310 IF A$="K" THEN GOSUB 950:PRINT "O.K VEEL SUCCES"
320 N=N+1:REM VOLGENDE KAART
330 GOSUB 930:GOSUB 1010
340 PRINT " DE ";K$
350 P(N)=V:SP=0
360 FOR L=1 TO N:SP=SP+P(L):NEXT L
370 IF SP>21 THEN PRINT "STUK.....":GOTO 410
380 IF N>4 THEN PRINT "VIJF KLEINTJES..."
390 IF SP=21 THEN PRINT "EEN EN TWINTIG..."
400 GOTO 260
410 REM NU DE COMPUTERS BEURT
420 PRINT "MIJN KAARTEN"
430 GOSUB 930
440 MT=0:GOSUB 1010:PRINT "EEN ";K$

```

```

450 M(1)=V:GOSUB 930:GOSUB 1010
460 PRINT "EEN ";K$:GOSUB 930:M(2)=V
470 IF M(1)=1 THEN M(1)=11
480 IF M(2)=1 AND M(1)<>11 THEN M(2)=11
490 IF M(1)+M(2)=21 THEN 620
500 IF P(1)+P(2)=21 AND N=2 THEN 660
510 MT=M(1)+M(2)
520 IF MT<SP AND SP<=21 THEN 550
530 PRINT "(";MT;")"
540 IF N<5 THEN PRINT "IK HOUD DEZE..":GOSUB 930:GOTO
  620
550 PRINT "IK NEEM NOG EEN KAART"
560 GOSUB 1010:GOSUB 930
570 PRINT " EEN ";K$
580 MT=MT+V:GOSUB 930
590 PRINT "(";MT;")"
600 IF MT>21 THEN PRINT "IK BEN STUK.....":GOTO 660
610 GOTO 520
620 REM SPELER VERLIEST
630 PRINT :PRINT "U VERLIEST DEZE KEER"
640 GOSUB 930:GOSUB 930
650 GOTO 100
660 REM COMPUTER VERLIEST
670 PRINT :PRINT "U WINT DEZE KEER"
680 GOSUB 930:GOSUB 930
690 M=M+(BET*2)
700 GOTO 100
710 REM ALS DE SPELER BLUT IS
720 PRINT :PRINT "U BENT BLUT"
730 GOSUB 930
740 PRINT "KOM TERUG OP UW GELUKSDAG"
750 GOSUB 930
760 END
770 REM KAARTEN SCHUDDEN
780 PRINT " ++ NIEUW PAK KAARTEN ++"
790 PRINT :PRINT "BEGIN-INZET IS 2 GULDEN"
800 PRINT :PRINT "EEN MOMENT A.U.B"
810 FOR L=1 TO 52
820 C(L)=L:NEXT L
830 REM SCHUDDEN
840 FOR L=1 TO 100
850 IF V>10 THEN V=10
860 A=INT(RND(1)*52+1)
870 B=INT(RND(1)*52+1)
880 T=C(A):C(A)=C(B)
890 C(B)=T
900 NEXT L
910 Q=1
920 RETURN
930 REM KORTE WACHT-ROUTINE

```

```
940 FOR WW=1 TO 100:NEXT WW:RETURN
950 REM INZET VERHOGEN
960 PRINT " UW INZET IS NU : ";BET
970 PRINT " U VERHOOGT UW INZET MET : ";
980 INPUT BB:IF M-BB<0 THEN PRINT "U HEEFT NIET ZOVEE
L GELD":GOTO 970
990 BET=BET+BB:M=M-BB
1000 RETURN
1010 REM KAART IN K
1020 REM NAAM IN K$
1030 K=C(Q):Q=Q+1:IF Q=53 THEN GOSUB 770:Q=1
1040 S=INT((K-1)/13):V=K-13*S:S=S+1
1050 ON V GOSUB 1090,1100,1110,1120,1130,1140,1150,11
60,1170,1180,1190,1200,1210
1060 ON S GOSUB 1220,1230,1240,1250
1070 IF V>10 THEN V=10
1080 RETURN
1090 K$="AAS ":RETURN
1100 K$="TWEЕ ":RETURN
1110 K$="DRIE ":RETURN
1120 K$="VIER ":RETURN
1130 K$="VIJF ":RETURN
1140 K$="ZES ":RETURN
1150 K$="ZEVEN ":RETURN
1160 K$="ACHT ":RETURN
1170 K$="NEGEN ":RETURN
1180 K$="TIEN ":RETURN
1190 K$="BOER ":RETURN
1200 K$="VROUW ":RETURN
1210 K$="HEER ":RETURN
1220 K$(LEN(K$)+1)="KLAVER ":RETURN
1230 K$(LEN(K$)+1)="RUITEN ":RETURN
1240 K$(LEN(K$)+1)="HARTEN ":RETURN
1250 K$(LEN(K$)+1)="SCHOPPEN ":RETURN
```

Tovenaars en heksen

Als u het spel kent van Kerkers en Draken, en dergelijke, dan heeft u een idee van het spel waarin u zich nu gaat bevinden, maar als u het nog niet kent dan kunnen we u beter een paar tips geven.

Als een goede tovenaars of tovenaarsster bevindt u zich op het zesde niveau van een onvoorstelbaar groot doolhof. U hoeft slechts naar de oppervlakte te ontsnappen. Gemakkelijk hè? Maar we hebben verschillende objecten op uw weg gelegd die u moet overwinnen met uw attributen.



Hoe te spelen

Om te ontsnappen, moet u zes etages naar boven, maar als u wilt, kunt u ook eerst naar beneden om uw score te verhogen, voordat u naar de oppervlakte gaat. U wordt gevraagd een keuze te maken gebaseerd op uw attributen, sterkte, gezondheid en toverkracht.

- Om te bewegen, dient u alleen de opgegeven toetsen in te drukken zonder op de RETURN-toets te drukken.
- We zouden het u niet moeten vertellen maar we denken dat u het beste van de monsters weg kunt rennen. Ze hebben de neiging gemeen te vechten en kunnen u verschikkelijk toetakelen.

Als u niet meer kunt ontsnappen druk dan, als laatste redmiddel, op de BREAK-toets om het spel te stoppen.

Programmeertips

U kunt de moeilijkheidsgraad veranderen door de waarden van uw 'KARAKTER-ATTRIBUTEN' in regel 90 te veranderen. Dit zal de onervaren speler helpen en hem de vaardigheid geven van Superman en hem langer laten leven. Als u echt een lafaard bent (wat we altijd al dachten) dan verandert u in regel 70, LV=6 in LV=1. Dat betekent dat u zult beginnen op moeilijkheidsgraad 1 en zonder te veel moeite kunt ontsnappen.

Lafhartige schurk, ga naar moeilijkheidsgraad 6 en vecht u een weg naar de uitgang.

Programma

```

10 REM TOVENAARS EN HEKSEN
20 DIM N$(80),MG(10),FI(10),ST(10)
30 DIM T$(10),E$(6),A$(70)
40 GOSUB 990
50 GRAPHICS 0
60 OPEN #1,4,0,"K:"
70 LV=6
80 SC=0
90 ST=10:HL=ST+5:MG=2
100 REM NU LOKATIE
110 E$(1,6)="      "
120 SC=SC+1
130 IF LV=0 THEN 880
140 M=0
150 MG=MG+1
160 IF HL<ST+5 THEN HL=HL+1
170 IF RND(1)>0.7 THEN E$(1)="N      "
180 IF RND(1)>0.7 THEN E$(2)="O      "
190 IF RND(1)>0.7 THEN E$(3)="Z      "
200 IF RND(1)>0.7 THEN E$(4)="W      "
210 IF RND(1)>0.95 THEN E$(5)="H      "
220 IF RND(1)>0.95 THEN E$(6)="L"
230 IF E$="      " THEN 170
240 PRINT "U BENT NU IN EEN NIEUWE KAMER"
250 PRINT :PRINT "DIT ZIJN UITGANGEN : "
260 PRINT "      ";E$
270 PRINT
280 IF RND(1)<0.7 THEN 320
290 M=1:A=INT(RND(1)*10+1):B=INT(RND(1)*10+1)
300 PRINT " ER IS EEN ";:X=A:GOSUB 1130
310 X=B:GOSUB 1170:PRINT " HIER"
320 PRINT :PRINT :PRINT
  
```

```

330 PRINT "WAT WILT U DOEN"
340 PRINT :PRINT "VECHTEN / AFWEREN / BEWEGEN / RENNE
N"
350 GET #1,T:T%=CHR$(T)
360 IF T%<>"V" AND T%<>"A" AND T%<>"B" AND T%<>"R" TH
EN PRINT " !!! PARDON !!!":PRINT :GOTO 320
370 IF T%="B" AND M=1 THEN PRINT "U KUNT HIER NIET WE
G":GOTO 320
380 IF T%="V" AND M=0 THEN PRINT "... ER VALT NIETS T
E VECHTEN":GOTO 320
390 IF T%="R" AND M=0 THEN PRINT "GOEDE OEFENING... D
ENK IK... ":GOTO 110
400 IF T%="R" AND RND(1)>0.6 THEN PRINT " U KWAM NIET
WEG ":GOTO 530
410 IF T%="R" THEN 110
420 IF T%="V" THEN PRINT "... WEES ABSOLUUT ZEKER..."
:PRINT "... HET IS UW IDEE ...":GOTO 530
430 IF T%="A" AND M=0 THEN PRINT "... GOEDE OEFENING
...":PRINT ".....NU.....":GOTO 320
440 IF T%="A" THEN PRINT "... WACHT EFFE JOCHIE":GOTO
530
450 PRINT :PRINT "WELKE RICHTING ";E%
460 PRINT "? ";
470 GET #1,T:T%=CHR$(T):PRINT T%:F=0:FOR L=1 TO 6
480 IF E$(L,L)=T% THEN F=1
490 NEXT L:IF F=0 THEN PRINT "...DIE KANT KUNT U NIET
OP...":GOTO 450
500 IF T%="H" THEN LV=LV-1:PRINT "ZO KOMT U HIER NOG
WEL EENS UIT..."
510 IF T%="L" THEN LV=LV+1:PRINT "LAGER EN LAGER GAAN
IS ":PRINT "SLECHT VOOR UW GEZONDHEID"
520 GOTO 110
530 REM VECHTEN MET HET MONSTER
540 S=INT(RND(1)*ST(B)+1):M=INT(RND(1)*MG(B)+1)
550 PRINT "TOVEREN / VECHTEN / RENNEN"
560 GET #1,T:T%=CHR$(T):IF T%<>"T" AND T%<>"V" AND T%
<>"R" THEN PRINT " ??? WAT ZEGT U ???":GOTO 550
570 IF T%="R" AND RND(1)>0.15 THEN PRINT "U KUNT HIER
NIET WEG":GOTO 630
580 IF T%="R" THEN 110
590 IF T%="T" AND MG<=0 THEN PRINT "U HEEFT AL UW MAG
IE GEBRUIKT":GOTO 630
600 IF T%="V" AND RND(1)>0.7 THEN PRINT "U MISTE":GOT
O 630
610 IF T%="V" THEN S=S-INT(RND(1)*ST+1):PRINT "EEN SL
AG... HOE HEFTIG":GOTO 630
620 T=INT(RND(1)*MG+1):S=S-T:MG=MG-T:PRINT "POW.,ZAP.
.WAT EEN GOEDE FLITS":GOTO 630
630 REM MONSTER AANVAL
640 IF S<=0 THEN PRINT "...U HEEFT DE ";:X=A:GOSUB 113

```

```

0:X=B:GOSUB 1170:PRINT :GOTO 730
650 T=INT(RND(1)*100):IF T>FI(B) AND M<>0 THEN PRINT
"WAT EEN TOVERKRACHT":GOTO 690
660 PRINT "DE VIJAND PROBEERT U TE SLAAN":IF RND(1)>0
.7 THEN PRINT "MAAR MIST":GOTO 710
670 PRINT "EN RAAKT U"
680 HL=HL-INT(RND(1)*S+1):PRINT "OH NEE, NOG MEER SCH
ADE":GOTO 700
690 T=INT(RND(1)*M+1):M=M-T:HL=HL-T
700 IF HL<=0 THEN 790
710 PRINT "U LEEFT NOG STEEDS!!":PRINT "DAT HAD IK NO
OIT GEDACHT"
720 GOTO 530
730 REM MONSTER IS DOOD
740 PRINT "GOED GEDAAN BRAVE RIDDER"
750 T=INT(RND(1)*(MG(B)*2+ST(B)))*5
760 PRINT "U HEEFT ";T;" STUKKEN GOUD"
770 SC=SC+T:M=0
780 GOTO 250
790 REM SPELER IS DOOD
800 PRINT :PRINT
810 PRINT "HET HOF ZAL OM U ROUWEN...."
820 PRINT "HET VOOREBIJGAAN VAN EEN GROOT KRIJGER"
830 PRINT "BEKEND OM ZIJN MOED EN STERKE WIL"
840 PRINT "(D.W.Z. U BENT NU IN HET HIERNAMAALS)"
850 FOR L=1 TO 600:NEXT L
860 PRINT "TROUWENS UW SCORE WAS :";SC
870 END
880 REM SPELER KWAM ER UIT
890 GRAPHICS 0:POKE 752,1:PRINT
900 POSITION 0,12:PRINT #6;"      U BENT ONTSNAPT"
910 FOR L=1 TO 255:POKE 710,L:POKE 712,(255-L)
920 NEXT L
930 PRINT "UW SCORE WAS:"
940 PRINT "      ";SC
950 PRINT "PLUS EEN KLEIN PSYCHOLOGISCH VOORDEEL"
960 FOR L=1 TO 500:NEXT L:PRINT :PRINT "...U LEEFT NO
G"
970 GET #1,A
980 END
990 REM MONSTERS INLEZEN
1000 RESTORE :N$=""
1010 FOR L=1 TO 10
1020 READ T$:N$(LEN(N$)+1)=T$
1030 NEXT L
1040 A$=""
1050 FOR L=1 TO 9
1060 READ T$:A$(LEN(A$)+1)=T$
1070 NEXT L
1080 FOR L=1 TO 10

```

```


1090 READ A,B,C
1100 MG(L)=A:ST(L)=B:FI(L)=C
1110 NEXT L
1120 RETURN
1130 REM BIJVOEGELIJKE NAAMWOORDEN
1140 Q=(X-1)*6+1
1150 PRINT A$(Q,Q+5);" ";
1160 RETURN
1170 REM PRINT DE MONSTERNAAM
1180 Q=(X-1)*7+1
1190 PRINT N$(Q,Q+6);" ";
1200 RETURN
1210 DATA ELF      ,DWERG  ,TROL   ,HEKS   ,PAD    ,HEK
S    ,ZOMBIE ,TOR    ,ENGERIK,CHIMERA
1220 DATA RUWE   ,WREDE  ,DOMME  ,VIEZE  ,GROTE  ,BRUINE,N
ATTE  ,ZWARTE,GEMENE
1230 DATA 5,5,10
1240 DATA 1,8,70
1250 DATA 0,15,100
1260 DATA 20,7,5
1270 DATA 0,20,100
1280 DATA 0,3,100
1290 DATA 3,10,65
1300 DATA 0,5,100
1310 DATA 2,8,50
1320 DATA 10,40,50

```

Woordpuzzel

Dit programma is ontworpen om de scherpzinnigheid te testen van degenen die denken dat ze goede ogen of een goed verstand hebben.

Het scherm zal bestaan uit een aantal letters waar woorden moeten uitkomen, woorden die u moet intoetsen. De woorden kunnen horizontaal, verticaal, diagonaal en zelfs achterwaarts worden ingevuld.



Hoe te spelen

Heeft u een woord gevonden, toets dan het antwoord in en druk op RETURN. U moet bedenken dat u misschien kleinere woordjes zult zien zoals 'OF' en 'EN' die snel zullen verschijnen vanwege de structuur van het scherm, maar deze zijn niet acceptabel voor de computer.

Programmeertips

Als u het programma wilt verbeteren, kunt u iemand anders de DATA-woorden laten intoetsen die aan het einde van het programma staan, anders weet u de woorden die worden gevraagd.

Programma

```

10 REM WOORDPUZZEL
20 DIM S$(225),N$(10),Q$(1)
30 DIM F(10),X(10),Y(10)
40 DIM A$(10)
50 DIM W(10)
60 GOSUB 380
70 GOSUB 660
80 FOR L=1 TO 10:W(L)=0:NEXT L
90 SC=0
100 GOTO 120
110 PRINT "WEES VERSTANDIG":FOR L=1 TO 100:NEXT L
120 PRINT "UW GOK";
130 INPUT A$
140 IF LEN(A$)=0 OR LEN(A$)>10 THEN 110
150 GOSUB 810
160 IF F=0 THEN PRINT "DIE IS ER NIET":GOTO 300
170 SC=SC+1
180 REM ACCENTUEER ONTDEKTE WOORDEN
190 IF W(F)=1 THEN PRINT "DIE IS AL GEBRUIKT!":GOTO 300
200 W(F)=1
210 B=INT(P(F)/15):A=P(F)-15*B
220 FOR L=1 TO 0
230 LOCATE A,B,D
240 POSITION A,B:PRINT #6;CHR$(D+32)
250 A=A+X(F):B=B+Y(F)
260 NEXT L
270 IF SC=10 THEN 320
280 PRINT "GOED GEDAAN ";SC
290 GOTO 300
300 FOR L=1 TO 50:NEXT L
310 GOTO 120
320 REM ALS ER 10 ZIJN GEVONDEN
330 PRINT :PRINT :PRINT
340 PRINT "U HEEFT ZE ALLEMAAL"
350 PRINT :PRINT "GOED GEDAAN"
360 FOR L=1 TO 500:NEXT L
370 END
380 REM DATA INLEZEN EN VIERKANT TEKENEN
390 GRAPHICS 1+16
400 POSITION 0,10:PRINT #6;" EEN MOMENT A.U.B."
410 S$="":FOR L=1 TO 15
420 S$(LEN(S$)+1)=" " :NEXT L
430 RESTORE 900
440 FOR I=1 TO 10
450 READ N$:N=LEN(N$)
460 ON INT(RND(1)*8+0.5) GOSUB 730,740,750,760,770,78

```

```


0,790,800
470 A=INT(RND(1)*15):B=INT(RND(1)*15)
480 C=A+X(I)*N:D=B+Y(I)*N
490 IF C<0 OR D<0 OR C>14 OR D>14 THEN 460
500 L=-1
510 L=L+1
520 Q=1+(B+Y(I)*L)*15+A+X(I)*L:Q$=S$(Q,Q)
530 IF Q$=" " THEN 550
540 IF Q$<>N$(L+1) THEN 460
550 IF L<>N-1 THEN 510
560 FOR L=0 TO N-1
570 Q=1+(B+Y(I)*L)*15+A+X(I)*L
580 S$(Q,Q)=N$(L+1)
590 NEXT L
600 F(I)=A+B*15
610 NEXT I
620 FOR L=1 TO 225
630 IF S$(L,L)=" " THEN S$(L,L)=CHR$(65+INT(RND(1)*26
))
640 NEXT L
650 RETURN
660 REM LAAT SCHERM ZIEN
670 GRAPHICS 1
680 FOR L=0 TO 14
690 POSITION 0,L
700 PRINT #6;S$(L*15+1,L*15+15)
710 NEXT L
720 RETURN
730 X(I)=-1:Y(I)=-1:RETURN
740 X(I)=-1:Y(I)=0:RETURN
750 X(I)=-1:Y(I)=1:RETURN
760 X(I)=1:Y(I)=-1:RETURN
770 X(I)=1:Y(I)=0:RETURN
780 X(I)=1:Y(I)=1:RETURN
790 X(I)=0:Y(I)=1:RETURN
800 X(I)=0:Y(I)=-1:RETURN
810 REM NAKIJKEN
820 RESTORE 900
830 F=0
840 FOR L=1 TO 10
850 READ N$
860 IF N$=A$ THEN F=L:O=LEN(N$)
870 NEXT L
880 RETURN
890 REM WOORDENSCHAT
900 DATA ATARI,VIDEO,KLEUR
910 DATA GELUID,GRAPHICS,RUIMTE
920 DATA TOEKOMST,PROGRAMMA,TOETSEN
930 DATA COMPUTER

```

Eindelijk, zegt u wellicht, in een computerboek moet ergens een letterwoord zijn, omdat de hele zaak vol zit met POKE, PEEK, BYTE en RAM. Nou, de titel betekent: Semi Intelligente Machinale Psycholoog met Uitbreidbare Respons.

Wat, zult u zeggen, heeft dit te maken met het normale dagelijkse leven. Dat zullen we even uitleggen.

Dit is uw kans om uw computer te gebruiken tegen uw ontroostbare vrienden, gasten, geliefden enz. Uw 'simpurtieke' ATARI zal vragen stellen en aandachtig reageren op uw karaktertrekken.



Hoe te spelen

Uw computer zal u verzoeken een beschrijving te geven van uzelf. Typ uw antwoorden in, maak ze zo lang mogelijk en uw ATARI XL computer zal zijn weloverwogen kritiek geven over uw karaktereigenschappen.

Het is alsof u een gratis abonnement heeft bij uw persoonlijke psycholoog.

Programmeertips

Mocht u het programma willen veranderen, dan kunt u het eenvoudigste meer DATA-regels toevoegen vanaf regel 590 en de computer een grotere keuze uit zijn antwoordenpakket geven.

Programma

```

10 REM SIMPUR
20 DIM Q$(200),A$(50)
30 DIM O$(100),T$(20):? CHR$(125)
40 Q$="HALLO, IK HEET SIMPUR, IK BEN HIER OM UW PROBL
EMEN TE BESPREKEN.          VERTEL IETS OVER UZELF.
"
50 BS=0
60 OS=0
70 ? Q$:? :GOTO 100
80 ? "KOM NOU, EEN BEETJE POSITIEF HE!":? :? :GOTO 10
0
90 ? :? Q$
100 INPUT A$:IF A$="" THEN 80
110 IF A$="JA" THEN ? "EINDELIJK EEN BEETJE POSITIEF"
:Q$="BENT U ALTIJD ZO DIREKT":GOTO 90
120 TRAP 90:IF A$="HOEZO" OR A$(1,2)="HE" OR A$="WAT"
THEN 90
130 IF A$="NEE" THEN ? "KORT EN KRACHTIG!":IF OS=0 TH
EN ? "GEEF ME IETS OM MEE TE WERKEN WILT U!":GOTO 90
140 IF A$(1,3)="IK " THEN A$=A$(4,LEN(A$))
150 IF A$(1,4)="JIJ " THEN A$=A$(4,LEN(A$))
160 A$(LEN(A$)+1)=" "
170 GOSUB 330:GOSUB 380
180 O$=A$:A$=" ":A$(2)=O$
190 IF WE=0 AND NA<>0 THEN GOSUB 440:GOTO 90
200 IF NA=0 AND WE<>0 THEN GOSUB 490:GOTO 90
210 IF WE=0 AND NA=0 THEN GOSUB 530:GOTO 90
220 IF WE<>0 AND NA<>0 THEN IF RND(0)>0.5 THEN GOSUB
440:GOTO 90
230 GOSUB 490:GOTO 90
240 REM ZOEK VOOR T$ IN A$ ROUTINE
250 F=0
260 IF A$=T$ THEN F=1:RETURN
270 IF LEN(A$)<LEN(T$) THEN RETURN
280 FOR L=2 TO LEN(A$)-LEN(T$)+1
290 IF A$(L-1,L-1)=" " THEN 320
300 IF A$(L-1,L+LEN(T$)-2)=T$ THEN F=1
310 REM : ? A$(L-1,L+LEN(T$)-2),T$,F
320 NEXT L:RETURN
330 REM GEEF WERKWOORD NR IN WE
340 WE=0:RESTORE 590
350 READ V:FOR Q=1 TO V
360 READ T$:GOSUB 240:IF F=1 THEN WE=Q:Q=V
370 NEXT Q:RESTORE 610:RETURN
380 REM GEEF NAAMWOORD NR IN NA
390 NA=0:REM RESTORE 600 IS AL GEDAAN
400 READ V:FOR Q=1 TO V

```



```
410 READ T$:GOSUB 240:IF F=1 THEN NA=Q:Q=V
420 NEXT Q:RETURN
430 REM ZOEK ANTWOORD OP NAAMWOORDEN
440 RESTORE 840:OS=1
450 FOR L=0 TO (NA-1)*3+INT(RND(1)*3)
460 READ Q$:NEXT L
470 RETURN
480 REM ZOEK ANTWOORD OP WERKWOORDEN
490 RESTORE 650:OS=1
500 FOR L=0 TO (WE-1)*3+INT(RND(1)*3)
510 READ Q$:NEXT L
520 RETURN
530 REM ZOEK ZOMAAR EEN ANTWOORD
540 OS=0:BS=BS+1:IF BS=7 THEN PRINT "SORRY, UW TIJD I
S OM !":PRINT CHR$(253):END
550 RESTORE 1030
560 FOR L=1 TO BS
570 READ Q$:NEXT L
580 RETURN
590 DATA 8
600 DATA HOU VAN,HAAT,HEB,WIL,VLIEG,LOOP,NEEM,GEEF
610 DATA 9
620 DATA AUTO,COMPUTER,VAKANTIE,GELD,TELEVISIE,MUZIEK
,VIDEO,ETEN
630 DATA VIS
640 REM ANTWOORD DATA
650 DATA HOE LANG WEET U AL DAT HET LIEFDE IS?
660 DATA HOE LANG KOESTERT U DEZE GEVOELEN AL?
670 DATA WANNEER IS DIT GEVOEL BEGONNEN?
680 DATA WEET U ZEKER DAT U NIET TE HAASTIG BENT,N
OU...ALS U HET ZEKER WEET DAN ZAL HET WEL GOED ZIJ
N !
690 DATA IS DIT EEN INTENS GEVOEL? MEENT
U HET OOK ECHT?
700 DATA DAT VIND IK ERG BEZITTERIG VAN U ! IS 'ZO
U GRAAG' NIET EEN BETERE UITDRUKKING ?
710 DATA MAAR BENT U WEL EIGENAAR VAN UW ZIEL..OF HEE
FT U HET GERUILD VOOR UW PLAATS IN DEZE CONSUMENTENWE
RELD?
720 DATA 'N BEETJE DUIDELIJKER AUB! HOU VAN.. OF WIL
LEN HEBBEN... DE TWEDE IS IETS BELEEFDER WEET U ?
730 DATA DA'S NOGAL IMPULSIEF !
740 DATA KINDEREN DIE WILLEN... DIE KRIJGEN VOOR H
UN BILLEN ! OF NIET ?
750 DATA WOU U OOK SOMS DAT U VLIEGEN KON ? MAAR U
LIEGTUIGEN ZIJN ZO DUUR TEGENWOORDIG... NIET
SOMS?
760 DATA VLIEG MEE... NAAR DE STILLE ZUIDZEE.. IS DAT
WAT U EIGENLIJK WILT?
770 DATA ER GAAT TOCH NIETS BOVEN DE BILLEWAGENOM VOO
```

RUIT TE KOMEN !
 780 DATA ZOU HET DAN SNELLER GAAN MET DE AUTO ?
 790 DATA MISSCHIEEN ALS U EERST ALSJEBLIEFT ZEI ... U
 WORDT WEER WAT TE INHALIG VOOR MIJ !
 800 DATA EN WAT VINDT U DAARVAN ?
 810 DATA GEVEN IS GOED VOOR UW KARAKTERVORMING.IK HOO
 P DAT U HET VEEL DOET !
 820 DATA MET KERST KUNT U GEVEN WAT U WILT !
 830 DATA WEES EERLIJK... KRIJGT U MEER DAN U GEEFT
 ?
 840 DATA HOUD U VAN SNELLE AUTO'S?,BENT U MONTEUR OF
 ZO!,HOUD U OOK ZO VAN SNELWEGEN? TE VEEL FILES VOLGE
 NS MIJ!
 850 DATA KOMPJOETER KRAZIE NOEMEN ZE DAT. BENT U
 ECHT ZO GOED MET DE TOETSEN ?
 860 DATA HEEFT U PROBLEMEN MET COMPUTERS OF ZO?VINDT
 U HET NIET VREEMD OM ZO MET EEN COMPUTER TE ZITTEN PR
 ATEN?
 870 DATA NEEMT U MIJ OOK MEE MET UW VAKANTIE, 'T IS
 ZO GOED OM ER EVEN TUSSENUIT TE ZIJN.. VINDT U OOK NI
 ET ?
 880 DATA IS HET LEUKER IN HET BUITENLAND ? OF BLI
 JFT U LIEVER DICHT BIJ HUIS ?
 890 DATA TJA...GELD IS IEDEREEN Z'N PROBLEEM...OVER G
 ELD TROUWENS... MIJN HONORARIUM LIEGT ER OOK NIET OM
 !
 900 DATA HOE VERDIENT U UW GELD. BENT U SOMS EEN WA
 ARZEGGER OF ZO ?
 910 DATA OH OM TOCH MILLIONAIR TE ZIJN... AL DAT
 KAVIAAR...KAN IK WEL VAN KOTSENWEET U !
 920 DATA IK MAAK ME ZORGEN OM OM AL DIE NIEUWE TELE-D
 IT EN TELE-DAT APPARATEN... U OOK ?
 930 DATA WAAR KIJKT U HET LIEFST TELEVISIE ?
 940 DATA BENT U NET ZO VERSLAAFD AAN DE T.V. ALS SO
 MMIGEN DIE IK KEN ?
 950 DATA VAN WELK SOORT MUZIEK HOUDT U ?,IK VOND VROE
 GER SIMON & GARFUNKEL LEUKHOE VINDT U ZE ?
 960 DATA ZO VEEL MODERNE MUZIEK IS TE HARD. IK KRI
 JG VEEL GEVALLEN HIER VAN DISCO-DOOFHEID.
 970 DATA GEBRUIKT U EEN VIDEORECORDER THUIS ? HOE LA
 NG HEEFT U HET AL DAN ?
 980 DATA ALS U TEVEEL VIDEO'T VERLIEST UW KONTAK
 T MET DE ECHTE WERELD WEET U !
 990 DATA DUS EET U TEVEEL ?,ETEN IS BEST MAAR VAN TE
 VEEL..... WORDT JE DIK VAN!!
 1000 DATA WAT IS UW FAVORIETE SCHOTEL DAN ?
 1010 DATA WILT U LIEVER OVER VISSSEN PRATEN DAN ?,EN V
 ISSSTICKS DAN ?
 1020 DATA HEEFT U WEL EENS EEN HELE DAG AAN EEN RIVIE
 R GEZETEN EN TOTAAL NIETS GEVANGEN ?



1030 DATA U HEBT ONGETWIFELD EEN HELEBOEL TALEN
TEN. SPEELT U OOK PIANO ?
1040 DATA WAT ZIJN ENKELE VAN UW HOBBIES DAN ?
1050 DATA WAAR WERKT U? OF ZIT U NOG OP SCHOOL?
1060 DATA EN NU EEN BEETJE WOORD-ASSOCIATIE... +++ W
ISKUNDE +++
1070 DATA WAT ZEGT U HET WOORD +++ MAKREEL +++
1080 DATA IK BEN BANG DAT WE ZO MOETEN OPHOUDEN. ER KO
MT ZO EEN HELE BELANGRIJKE PATIENT.

Catalogus

De meeste computerboeken zijn geschreven voor jongens en mannen die spelletjes willen spelen, maar er is niets bruikbaar voor meisjes en huisvrouwen. We denken echter dat vele programma's in dit boek voor iedereen in het gezin bruikbaar zijn. Dit programma is in ieder geval voor **IEDEREEN**. U kunt met dit programma alles opslaan: recepten, platen, telefoonnummers, voetbalstanden, boodschappenbriefjes en ga zo maar door.



Hoe te gebruiken

U wordt gevraagd welke keuze u wilt maken uit een 'menu'. Maak uw keuze. U hoeft niet nog een toets in te drukken daar de computer meteen reageert op uw eerste toetsaanslag.

Gaat u verder, druk dan op de RETURN-toets.

Onthoud dat de SPATIEBALK de gegevens zal 'WEGSCHRIJVEN' of 'INLADEN'.

Programmeertips

Als u een diskdrive bezit, dient u de twee OPEN #2 regels als volgt te veranderen:

OPEN #2,4,0, 'C:' wordt dan OPEN #2,4,0, 'D:DATA'

en

OPEN #2,8,0, 'C:' wordt dan OPEN #2,8,0, 'D:DATA'

Programma

```

10 REM KATALOGUS
20 OPEN #1,4,0,"K:"
30 DIM R$(500),I$(2000)
40 DIM S$(11),T$(41)
50 C=0
60 REM MENU
100 GRAPHICS 0:PRINT :PRINT "          KATALOGUS PROGRA
MMA"
110 PRINT "          ====="
120 PRINT :PRINT :PRINT
130 PRINT "1. DATA LADEN VAN CASSETTE":PRINT
140 PRINT "2. GEGEVENS WISSEN UIT GEHEUGEN":PRINT
150 PRINT "3. DATA SCHRIJVEN NAAR CASSETTE":PRINT
160 PRINT "4. GEGEVENS ZOEKEN":PRINT
170 PRINT "5. GEGEVENS INVOEREN"
180 PRINT :PRINT :PRINT
190 PRINT "  UW KEUZE ";
200 GET #1,T:IF T<49 OR T>53 THEN 200
210 ON T-48 GOSUB 1000,2000,3000,4000,5000
220 GOTO 100
1000 OPEN #2,4,0,"C:"
1010 INPUT #2,C
1020 FOR L=1 TO C
1030 T=1+(L-1)*10
1040 I=1+(L-1)*40
1050 INPUT #1,S$,T$
1060 R$(T,T+9)=S$:I$(I,I+39)=T$
1070 NEXT L
1080 CLOSE #2:RETURN
2000 PRINT " BENT U ZEKER (J/N)"
2010 GET #1,T:T$=CHR$(T):IF T$<>"J" THEN RETURN
2020 C=0:RETURN
3000 IF C=0 THEN RETURN
3010 OPEN #2,8,0,"C:"
3020 PRINT #2,C
3030 FOR L=1 TO C
3040 T=1+(L-1)*10
3050 I=1+(L-1)*40
3060 S$=R$(T,T+9):T$=I$(I,I+39)
3070 PRINT #2,S$,T$
3080 NEXT L
3090 CLOSE #2:RETURN
4000 IF C=0 THEN RETURN
4010 PRINT "NAAM WAARONDER GEZOCHT DIENT          TE
WORDEN :";
4020 INPUT S$:IF S$="" THEN 4010
4030 FOR L=1 TO C
4040 T=1+(L-1)*10

```

```

4050 I=1+(L-1)*40
4060 IF R$(T,T+LEN(S$)-1)=S$ THEN PRINT R$(T,T+9):PRI
NT I$(I,I+39)
4070 NEXT L
4080 GET #1,T
4090 RETURN
5000 C=C+1:IF C=51 THEN C=50:RETURN
5010 PRINT "RUBRIEK WAARONDER DIT ONTHOUDEN DIEN
T TE
WORDEN: ";
5020 INPUT S$:IF S$="" THEN 5010
5030 IF S$="STOP" THEN C=C-1:RETURN
5040 INPUT T$:IF T$="" THEN 5010
5050 IF T$="STOP" THEN C=C-1:RETURN
5060 S$(LEN(S$)+1)="      "
5070 T$(LEN(T$)+1)="      "
      "
5080 T=1+(C-1)*10
5090 I=1+(C-1)*40
5100 R$(T,T+9)=S$:I$(I,I+39)=T$
5110 GOTO 5000


```

Ruimtestad

Terug uit de ruimte met uw shuttle moet u landen met precies de goede snelheid en op exact de goede plaats in het midden van de stad.

Gaat u éven verkeerd dan zult u een van de hoge torens raken. Een te harde landing en u zult op de speciaal klaargemaakte landingsbaan ontploffen. Een even groot probleem krijgt u als u denkt dat u alles kunt, u zult dan in de ruimte blijven hangen ... voor altijd.

Trouwens, u heeft maar een bepaalde hoeveelheid brandstof, dus als u niet in een bepaalde tijd binnen bent, zult u onbestuurbaar in de stad terechtkomen.



Hoe te spelen

Om uw linkse en rechtse raketten te gebruiken, dient u op Z en X te drukken. Gebruik / en + om te stijgen of te dalen met behulp van de hoofdmotoren en u bent verzekerd van een goede landing.

Er zijn verschillende moeilijkheidsgraden. Als u wordt afgeschoten naar de ruimte zult u terugkeren in dezelfde moeilijkheidsgraad.

Programmeertips

Om het spel moeilijker of gemakkelijker te maken, kunt u de hoeveelheid brandstof dat u tot uw beschikking heeft, veranderen door de waarde van variabele F in regel 90 te veranderen.

Programma

```

10 REM RUIMTESTAD
20 DIM K(7)
30 G=5:REM ZHAARTEKRACHT
40 GRAPHICS 5+16
50 GOSUB 610
60 POKE C1,23
70 GOSUB 720
80 H=40:REM HOOGTE
90 F=300:REM BRANDSTOF
100 V=2:REM SNELHEID
110 T=3:REM STUKKRACHT
120 X=100:REM HORIZ. POSITIE
130 S=0:REM ZIJWAARTSE SNELHEID
140 REM HOOFD AFDALINGS LUS
150 SOUND 0,50,6,T+5
160 F=F-T:IF F<=0 THEN F=0:T=0:SETCOLOR 4,3,2
170 A=X:G=H
180 V=V+(G-T)/10
190 H=H+V:X=X+S
200 IF X<30 OR X>210 THEN 490
210 IF H<20 THEN 540
220 POKE 53278,1
230 GOSUB 860
240 IF PEEK(53252)<>0 THEN 380
250 IF PEEK(764)=23 AND S>-3 THEN S=S-1:POKE 764,0
260 IF PEEK(764)=22 AND S<3 THEN S=S+1:POKE 764,0
270 IF PEEK(764)=6 AND T<9 THEN T=T+1:POKE 764,0
280 IF PEEK(764)=38 AND T>0 THEN T=T-1:POKE 764,0
290 GOTO 140
300 REM EEN GEBOUW GERAAKT
310 FOR L=1 TO 255 STEP 3:POKE C1,255-L:POKE 712,L:SO
UND 0,L,4,14:NEXT L:SOUND 0,0,0,0
320 POKE H1,0
330 GRAPHICS 1+16:POSITION 0,10:PRINT #6;"      VERONGE
LUKT!"
340 POSITION 0,4:PRINT #6;"      BRANDSTOF: ";F
350 POSITION 0,14:PRINT #6;"      PROBEER IETS      M
AKKELIJKERS"
360 IF G<1 THEN G=1
370 GOTO 470
380 REM SCHIP IETS GEVONDEN
390 FOR L=1 TO 300:NEXT L
400 IF PEEK(53252)=5 OR PEEK(53252)=4 THEN 300
410 IF V>2 THEN 310
420 POKE H1,0
430 GRAPHICS 1+16:POSITION 0,10:PRINT #6;"      goed ge
land":POKE 53248,0
440 POSITION 0,4:PRINT #6;"      BRANDSTOF: ";F

```



```
450 POSITION 0,14:PRINT #6;" PROBEER IETS  
MOEILIJKERS"  
460 IF G>8 THEN G=8  
470 FOR L=1 TO 999:NEXT L  
480 GOTO 40  
490 REM PLANEET GEMIST  
500 POKE H1,0  
510 GRAPHICS 1+16:POSITION 0,10  
520 PRINT #6;" u miste het"  
530 GOTO 570  
540 REM NAAR DE RUIMTE GESCHOTEN  
550 GRAPHICS 1+16:POSITION 0,10  
560 PRINT #6;"U BENT IN EEN OMLOOPBAAN":POKE 53248,0  
570 FOR L=1 TO 500:NEXT L  
580 POSITION 0,15:PRINT #6;"ZELFDE MOEILIJKHEIDS  
GRAAD"  
590 FOR L=1 TO 500:NEXT L  
600 GOTO 40  
610 REM LANDER OPZETTEN  
620 H1=53248:C1=704:POKE H1,0  
630 POKE 559,62:POKE 53277,3  
640 PMBASE=PEEK(106)-24:POKE 54279,PMBASE  
650 P1=PMBASE*256+1024  
660 FOR I=0 TO 255:POKE P1+I,0:NEXT I  
670 RESTORE 930  
680 FOR I=0 TO 7:READ D  
690 K(I)=D  
700 POKE P1+I+40,D:NEXT I  
710 RETURN  
720 REM STAD EN GEBIED TEKENEN  
730 COLOR 3  
740 FOR L=0 TO 75 STEP 5  
750 H=INT(RND(1)*10+4)  
760 FOR I=0 TO 4:PLOT L+I,47  
770 DRAWTO L+I,47-H:NEXT I  
780 NEXT L  
790 COLOR 1:L=INT(RND(1)*14)*5  
800 FOR I=0 TO 4:PLOT L+I,47  
810 PLOT L+I,46:NEXT I  
820 COLOR 4  
830 FOR I=45 TO 31 STEP -1:PLOT L,I  
840 DRAWTO L+4,I:NEXT I  
850 RETURN  
860 REM WISSEN EN OVERNIEUW TEKENEN  
870 FOR I=INT(B) TO INT(B+7):POKE P1+I,0:NEXT I  
880 FOR I=0 TO 7:POKE P1+I+INT(H),K(I)  
890 NEXT I  
900 POKE H1,X  
910 RETURN  
920 REM LANDER MAKEN  
930 DATA 90,36,36,60,102,90,66,195
```

Dobbelen

Als u moe bent van het neerschieten van kernkoppen en het verdedigen van werelddelen dan is hier een programma om uw goklust op uit te leven.

De computer zal vijf dobbelstenen 'gooien' en u moet proberen de best mogelijke reeks te halen voor de hoogste score.

Mogelijkheden voor de te halen scores zijn:

Een paar

Drie gelijk

Twee paren


Drie gelijk en een paar

Vier gelijk

Kleine straat

Grote straat

Vijf gelijk



Hoe te spelen

Na de eerste 'worp' van de computer kunt u de ronde houden met de cijfers 1 t/m 5. Druk op de R om weg te geven.

Gebruik de SPATIEBALK om de ronde vrij te gooien. U heeft tien ronden om de hoogste score te krijgen en een eventuele over-de-schouder-kijker een doel te geven om op te richten.

Programma



```
10 REM DOBBELEN
20 GRAPHICS 1+16
30 DIM D(5),F(5),S(6),N$(20)
40 T=10
100 T=T-1:IF T=0 THEN 2000
110 FOR L=1 TO 5:F(L)=0:NEXT L
115 POSITION 0,5:PRINT #6;"
120 POSITION 0,5:PRINT #6;"
130 GOSUB 1000
140 GOSUB 1200
150 GOSUB 1000
160 GOSUB 1200
170 GOSUB 1000
180 GOSUB 3000
190 SC=SC+NS
200 GOTO 100
1000 REM MAAK EN TOON 5 DOBBELSTENEN
1010 FOR L=1 TO 5
1020 IF F(L)=1 THEN 1110
1030 Q=INT(RND(1)*60+10)
1040 D(L)=1
1050 FOR E=1 TO Q
1060 POSITION L*3,4
1070 D(L)=D(L)+1
1080 IF D(L)=7 THEN D(L)=1
1090 PRINT #6;D(L)
1100 NEXT E
1110 NEXT L
1120 RETURN
1200 REM STOPEKENS PRINTEN
1210 POSITION 4,1:PRINT #6;"VASTHOUDEN?"
1220 POSITION 0,5:PRINT #6;"
1230 FOR L=1 TO 5:F(L)=0:NEXT L
1240 OPEN #1,4,0,"K:"
1250 GET #1,A
1260 IF A=ASC(" ") THEN 1310
1270 IF A=ASC("R") THEN CLOSE #1:GOTO 1220
1280 IF A<ASC("1") OR A>ASC("5") THEN 1250
1290 POSITION 3*(A-48),5:PRINT #6;CHR$(ASC("+")+128)
1300 F(A-48)=1:GOTO 1250
1310 CLOSE #1
1320 POSITION 4,1:PRINT #6;"
1330 RETURN
2000 REM EINDE VAN HET SPEL
2010 GRAPHICS 1+16
2020 POSITION 1,10
2030 PRINT #6;"EINDE VAN HET SPEL"
```

```

2040 POSITION 4,13
2050 PRINT #6;"uw score is"
2060 POSITION 8,16
2070 PRINT #6;STR$(SC)
2080 POSITION 5,22
2090 PRINT #6;"<SPATIE>"
2100 OPEN #1,4,0,"K:"
2110 GET #1,A
2120 IF A<>ASC(" ") THEN 2110
2130 CLOSE #1
2140 RUN
3000 REM SCORE IN NS EN TYPE IN N$
3010 FOR L=1 TO 6:S(L)=0:NEXT L
3020 FOR L=1 TO 5
3030 S(D(L))=S(D(L))+1
3040 NEXT L
3050 Z2=0:FOR L=1 TO 6
3060 IF S(L)=1 THEN Z2=Z2+1
3070 NEXT L
3100 Z1=0:FOR L=1 TO 6
3110 IF S(L)=0 THEN Z1=Z1+1
3120 NEXT L
3130 IF Z1=5 THEN GOSUB 4000:GOTO 3500
3140 IF Z1=1 AND S(6)=0 THEN GOSUB 4350:GOTO 3500
3150 IF Z1=1 AND S(1)=0 THEN GOSUB 4400:GOTO 3500
3160 IF Z1=1 THEN GOSUB 4500:GOTO 3500
3170 IF Z1=2 THEN GOSUB 4300:GOTO 3500
3180 IF Z1<>3 THEN 3210
3190 IF Z2=1 THEN GOSUB 4100:GOTO 3500
3200 IF Z2=2 THEN GOSUB 4250:GOTO 3500
3210 IF Z2=1 THEN GOSUB 4200:GOTO 3500
3220 GOSUB 4050:GOTO 3500
3450 PRINT #6;" "
3500 REM SCORE ZIT NU IN NS
3510 REM BOODSCHAP ZIT NU IN N$
3530 POSITION 10-(INT(LEN(N$)/2)),10:PRINT #6;N$
3540 POSITION 6,14:PRINT #6;"SCORE :";STR$(NS);"
"
3550 POSITION 6,19:PRINT #6;"<SPATIE>"
3600 OPEN #1,4,0,"K:"
3610 GET #1,A
3620 IF A<>ASC(" ") THEN 3610
3630 CLOSE #1
3640 FOR Q=10 TO 19:POSITION 0,Q
3650 PRINT #6;" "
3660 NEXT Q
3670 RETURN
4000 REM VIJF GELIJK
4010 N$="***** EUREKA *****":NS=50:RETURN
4050 REM EEN PAAR EN DRIE


```

```
4060 N$="* 2 + 3 IS VIEREN":NS=30:RETURN
4100 REM TWEE PAREN
4110 N$="** DUBBEL DUBBEL **":NS=20:RETURN
4200 REM VIER GELIJK
4210 N$="**** VIER KEER ****":NS=15:RETURN
4250 REM THREE OF A KIND
4260 N$="*** TRIO ***":NS=10:RETURN
4300 REM EEN PAIR
4310 N$="**** DUBBEL ****":NS=7:RETURN
4350 REM KLEINE STRAAT
4360 N$="***** KLEINTJE *****":NS=24:RETURN
4400 REM GROTE STRAAT
4410 N$="***** 2+3+4+5+6 *****":NS=40:RETURN
4500 REM GEEN SCORE
4510 N$="***** NIX *****":NS=0:RETURN
```

Missile Command

Het onmogelijke is gebeurd.

Het verdedigingsleger is naar het heelal waar nog niemand is geweest en u bent de enige die drie steden kan verdedigen tegen een regen van voorwerpen. Uw verdedigingsveld is uitgeschakeld, dus u dient de wezens te stoppen voordat ze de onverdedigde steden bereiken. Ook al stopt u de eerste groep, ze blijven komen en de kleuren van het scherm zullen veranderen om het nog moeilijker te maken hun weg te voorspellen.



Hoe te spelen

De projectielen komen vanaf boven en gaan hun weg naar u. U moet hun baan kruisen en vuren met de SPATIEBALK. Onthoud dat uw kernkoppen er een seconde over doen om hun doel te bereiken, dus u dient eerder te schieten. Gebruik de Z en X om links en rechts te bewegen en / en + voor op en neer. Bij een score van 5000 of een veelvoud hiervan krijgt u een bonus per overgebleven stad. Omdat de vijanden verschillende waarden hebben, afhankelijk van de moeilijkheidsgraad, moet u af en toe besluiten één van uw steden te laten vernietigen om zo de 5000 punten bij elkaar te krijgen om verder te kunnen spelen. Dit lijkt misschien een tegenstelling, maar het gaat om de hoge score, nietwaar?

Programmeertips

Om de vijand minder goed te laten spelen, kunt u een pauze inlassen. Dit dient als volgt te gebeuren.

```
100 FOR QW = 1 TO 30 : NEXT QW : IF J = 0 THEN 110
```

Programma



```
10 CI=3:BO=0:SC=0:WV=1
20 DIM X(10),Y(10),A(10),B(10)
30 GRAPHICS 7+16:POKE 752,1
40 GOSUB 1010
50 N=INT(RND(1)*8+3)
60 GOSUB 700
70 C=0:M=N
80 IF WV<>1 THEN GOSUB 1240
90 IF CI<3 AND BO<>0 THEN CI=CI+1:BO=BO-1:GOTO 90
100 IF J=0 THEN 120
110 J=J-1:IF J=0 THEN E=1
120 C=C+1:IF C=80 THEN 510
130 IF N=0 THEN 510
140 FOR L=1 TO M
150 GOSUB 310
160 IF E=0 THEN 190
170 COLOR 2:S=E:X=Q:Y=R:SOUND 0,E*12,4,14:GOSUB 800:E
=E+1
180 IF E=21 THEN E=0:SOUND 0,0,0,0
190 IF Y(L)=-1 THEN 290
200 COLOR 3
210 PLOT X(L),Y(L)
220 X(L)=X(L)+A(L):Y(L)=Y(L)+B(L)
230 IF Y(L)>78 THEN Y(L)=-1:N=N-1:GOTO 290
240 LOCATE X(L),Y(L),D
250 IF D<>2 THEN 290
260 IF Y(L)>70 THEN GOSUB 1200
270 Y(L)=-1:N=N-1
280 SC=SC+10*WV
290 NEXT L
300 GOTO 370
310 IF PEEK(764)=33 AND E=0 AND J=0 THEN J=3:Q=G:R=H
320 IF PEEK(764)=23 AND G>5 THEN COLOR 4:PLOT G,H:COL
OR 2:G=G-1:PLOT G,H
330 IF PEEK(764)=6 AND H>5 THEN COLOR 4:PLOT G,H:COLO
R 2:H=H-1:PLOT G,H
340 IF PEEK(764)=22 AND G<154 THEN COLOR 4:PLOT G,H:CO
LOR 2:G=G+1:PLOT G,H
350 IF PEEK(764)=38 AND H<64 THEN COLOR 4:PLOT G,H:CO
LOR 2:H=H+1:PLOT G,H
360 RETURN
370 IF CI+BO>0 THEN 100
380 GRAPHICS 17
390 POSITION 0,5
400 PRINT #6;" score :";SC
410 POSITION 0,12
420 PRINT #6;" EINDE SPEL"
```

```

430 POSITION 0,7
440 PRINT #6;" OPDRACHT:";WV
450 POSITION 0,19
460 PRINT #6;" < RETURN >"
470 OPEN #1,4,0,"K:"
480 GET #1,D
490 IF D<>155 THEN 480
500 CLOSE #1:RUN
510 GRAPHICS 17
520 SOUND 0,0,0,0:SOUND 1,0,0,0
530 POSITION 0,5:PRINT #6;"EINDE VAN OPDRACHT:";WV
540 WV=WV+1
550 POSITION 0,10:PRINT #6;"          ";CI;" STEDEN"
560 POSITION 0,17:PRINT #6;"          bonus"
570 B=(CI+BO)*100*WV
580 FOR L=B TO 0 STEP -10
590 SOUND 1,L*200/B+40,10,10
600 POSITION 8,19:PRINT #6;" ";L;" "
610 NEXT L
620 SOUND 1,0,0,0
630 SC=SC+B:IF INT(SC/5000)=SC/5000 THEN BO=BO+1:POSITION 6,2:PRINT "BONUS STAD"
640 FOR L=1 TO 800:NEXT L:GOTO 30
650 COLOR 1
660 FOR S=1 TO 20
670 SOUND 0,S*12,4,14:GOSUB 800
680 NEXT S
690 SOUND 0,0,0,0:RETURN
700 FOR L=1 TO 10
710 X(L)=INT(RND(1)*160):Y(L)=0
720 S=INT(RND(1)*40)+40
730 ON INT(RND(1)*3+1) GOTO 740,750,760
740 A(L)=(25-X(L))/S:B(L)=(78-Y(L))/S:GOTO 770
750 A(L)=(65-X(L))/S:B(L)=(77-Y(L))/S:GOTO 770
760 A(L)=(135-X(L))/S:B(L)=(75-Y(L))/S
770 NEXT L
780 G=80:H=50:COLOR 1:PLOT G,H
790 J=0:E=0:RETURN
800 SS=S:IF S>10 THEN COLOR 4:SS=S-10
810 ON SS GOTO 820,830,840,850,860,880,900,920,940,970
820 PLOT X,Y:PLOT X+1,Y:PLOT X,Y+1:RETURN
830 PLOT X-1,Y:PLOT X,Y-1:RETURN
840 PLOT X+2,Y:PLOT X+1,Y+1:PLOT X,Y+2:PLOT X-1,Y+1:RETURN
850 PLOT X-2,Y:PLOT X-1,Y-1:PLOT X,Y-2:PLOT X+1,Y-1:RETURN
860 PLOT X-2,Y+2:PLOT X-2,Y+1:PLOT X-3,Y:PLOT X-2,Y-1:PLOT X-2,Y-2:PLOT X-1,Y-2
870 PLOT X,Y-3:PLOT X+1,Y-3:PLOT X+1,Y-2:RETURN

```



```
880 PLOT X+2,Y-2:PLOT X+2,Y-1:PLOT X+3,Y:PLOT X+2,Y+1
:PLOT X+2,Y+2:PLOT X+1,Y+2
890 PLOT X,Y+3:PLOT X-1,Y+3:PLOT X-1,Y+2:RETURN
900 PLOT X,Y-4:PLOT X-1,Y-4:PLOT X-1,Y-3:PLOT X-2,Y-3
:PLOT X-3,Y-2:PLOT X-3,Y-1:PLOT X-4,Y-1
910 PLOT X-4,Y:PLOT X-3,Y+1:PLOT X-3,Y+2:PLOT X-2,Y+3
:PLOT X-1,Y+4:RETURN
920 PLOT X,Y+4:PLOT X+1,Y+4:PLOT X+1,Y+3:PLOT X+2,Y+3
:PLOT X+2,Y+2:PLOT X+3,Y+1:PLOT X+4,Y+1
930 PLOT X+4,Y:PLOT X+3,Y-1:PLOT X+3,Y-2:PLOT X+2,Y-3
:PLOT X+1,Y-4:RETURN
940 PLOT X,Y-5:PLOT X-1,Y-5:PLOT X-2,Y-4:PLOT X-3,Y-3
:PLOT X-4,Y-2:PLOT X-5,Y-1:PLOT X-5,Y
950 PLOT X-5,Y+1:PLOT X-4,Y+1:PLOT X-4,Y+2:PLOT X-3,Y
+3:PLOT X-2,Y+4:PLOT X-1,Y+5
960 RETURN
970 PLOT X,Y-5:PLOT X+1,Y+5:PLOT X+2,Y+4:PLOT X+3,Y+3
:PLOT X+4,Y+2:PLOT X+5,Y+1:PLOT X+5,Y
980 PLOT X+5,Y-1:PLOT X+4,Y-1:PLOT X+4,Y-2:PLOT X+3,Y
-3:PLOT X+2,Y-4:PLOT X+1,Y-5
990 RETURN
1000 GOTO 1000
1010 COLOR 1
1020 RESTORE 1340
1030 READ D:PLOT 0,D
1040 FOR L=10 TO 150 STEP 10
1050 READ D:DRAWTO L,D:NEXT L
1060 READ D:DRAWTO 159,D
1070 COLOR 2
1080 ON CI GOTO 1110,1100,1090
1090 X=20:Y=78:GOSUB 1130
1100 X=60:Y=77:GOSUB 1130
1110 X=130:Y=75:GOSUB 1130
1120 RETURN
1130 RESTORE 1350
1140 PLOT X,Y
1150 READ A,B
1160 DRAWTO X+A,Y+B
1170 X=X+A:Y=Y+B
1180 IF A<>0 OR B<>0 THEN 1150
1190 RETURN
1200 IF X(L)<35 THEN X=25:Y=78:GOSUB 650:CI=CI-1:X=20
:COLOR 4:GOSUB 1130:GOTO 1230
1210 IF X(L)<75 THEN X=65:Y=77:GOSUB 650:CI=CI-1:X=20
:COLOR 4:GOSUB 1130:GOTO 1230
1220 X=135:Y=75:GOSUB 650:CI=CI-1:X=130:COLOR 4:GOSUB
1130:GOTO 1230
1230 RETURN
1240 C1=INT(RND(1)*16)
1250 C2=INT(RND(1)*16):IF C2=C1 THEN 1250
```

```
1260 C3=INT(RND(1)*16):IF C3=C2 OR C3=C1 THEN 1260
1270 C4=INT(RND(1)*16):IF C4=C3 OR C4=C2 OR C4=C1 THE
N 1270
1280 C5=INT(RND(1)*16):IF C5=C4 OR C5=C3 OR C5=C2 OR
C5=C1 THEN 1280
1290 SETCOLOR 1,C2,INT(RND(1)*10+5)
1300 SETCOLOR 2,C3,INT(RND(1)*10+5)
1310 SETCOLOR 3,C4,INT(RND(1)*10+5)
1320 SETCOLOR 4,C5,INT(RND(1)*16)
1330 RETURN
1340 DATA 72,75,79,79,73,76,78,78,74,71,72,74,79,76,7
6,73,70
1350 DATA 9,0,0,-3,-1,0,0,-2,-3,0,0,3
1360 DATA -2,0,0,-2,-2,0,0,3,-1,0,0,1
1370 DATA 3,0,0,-2,4,0,0,1,1,0,0,-1
1380 DATA 0,0
```

BIORITME ... Grafieken en voorspelling

Maakt u zich geen zorgen, het doet geen pijn, gaat u alleen even liggen en geniet van de sensatie en uw leven dat voorbijflitst.

In simpele termen is dit een systeem dat zal voorspellen of kijken hoe uw dag wordt en of uw dag morgen zo slecht wordt als u denkt. U typt de datum in en uw computer zal u vertellen en laten zien hoe uw dag wordt. Het geeft ook een idee van beide kanten van de dag waar we nu zijn.



Hoe te gebruiken

De computer zal u vragen uw geboortedatum in te toetsen en de dag waarop u het programma gebruikt.

Er zal worden gevraagd naar het jaar, de maand en de dag of de week die u wilt laten voorspellen. Als u wordt gevraagd naar de dag, denk er dan aan dat u het dagnummer moet intypen, omdat er meer 'dinsdagen' in een maand zijn.

Druk op RETURN na elke invoer.

Nadat alles is ingetoetst, zal de computer een kaart laten zien van uw:

- Lichamelijke
 - Emotionele
 - en
 - Intellectuele
- gesteldheid van de dag.

Zoals we reeds hebben gezegd, laten we een dag ervoor en een dag erna zien, en kunt u alsnog besluiten niet uit uw bed te komen voor de komende week.

Druk op RETURN voor het menu.

Programma



```
10 REM BIORITME
12 DEG
15 DIM T$(20),M$(20)
20 GRAPHICS 0:POSITION 0,4:PRINT "GEBORTE DATUM : "
30 GOSUB 1000
40 BJ=J:BM=M:BD=D
50 GRAPHICS 0:POSITION 0,4:PRINT "DATUM TE ONDERZOEKE
N : "
60 GOSUB 1000
100 GOSUB 2000
110 LICH=DAGEN-23*INT(DAGEN/23)
120 EMOT=DAGEN-28*INT(DAGEN/28)
130 ITEL=DAGEN-33*INT(DAGEN/33)
140 GRAPHICS 7:POKE 752,1:PRINT
150 PRINT "LICHAMELIJKE CYCLUS (ROOD)"
160 QQ=LICH/23*360:MU=28/23
170 COLOR 1:GOSUB 3000
200 PRINT "EMOTIONELE CYCLUS (GROEN)"
210 QQ=EMOT/28*360:MU=1
220 COLOR 2:GOSUB 3000
300 PRINT "INTELLECTUELE CYCLUS (BLAUW)"
310 QQ=ITEL/33*360:MU=28/33
320 COLOR 3:GOSUB 3000
330 PRINT "          DATUM OPGEGEVEN"
340 PLOT 80,0:DRAWTO 80,79
400 END
410 REM *****
1000 REM INLEZEN VAN EEN DATUM
1010 POSITION 0,10:PRINT "JAAR.....
      ":POSITION 9,10
1020 INPUT J:IF J<>INT(J) THEN 1010
1030 POSITION 0,15:PRINT "MAAND....
      ":POSITION 9,15
1040 INPUT M$
1050 IF M$(1,1)>"0" AND M$(1,1)<="9" THEN M=VAL(M$):G
OTO 1100
1060 RESTORE 4000
1070 M=0:FOR L=1 TO 12
1080 READ T$:IF T$(1,3)=M$(1,3) THEN M=L
1090 NEXT L
1100 IF M=0 OR M<>INT(M) OR M>12 THEN 1030
1110 POSITION 0,20:PRINT "DAG.....
      ":POSITION 9,20
1120 INPUT D:IF D<1 OR D>31 OR D<>INT(D) THEN 1110
1130 RETURN
1140 REM *****
2000 REM AANTAL DAGEN BEREKENEN
```

```

2010 DP=0:DM=0
2020 DP=365.25*J:DM=365.25*BJ
2030 RESTORE 4500
2040 FOR L=1 TO 12
2050 READ T
2060 IF M=L THEN DP=DP+T
2070 IF BM=L THEN DM=DM+T
2080 NEXT L
2090 DP=DP+D:DM=DM+BD
2100 DAGEN=DP-DM
2200 RETURN
2210 REM *****
3000 REM TEKEN EEN CYCLUS
3010 PLOT 0,75-(37.5+37.5*SIN(-360*MU+QQ))
3020 FOR TH=-360 TO 350 STEP 2.5
3030 X=TH/4.5+80:J=37.5+37.5*SIN(TH*MU+QQ)
3040 DRAWTO X,75-J
3050 NEXT TH
3060 FOR W=1 TO 600:NEXT W
3070 RETURN
3080 REM *****
4000 DATA JAN,FEB,MAA,APR,MEI,JUN
4010 DATA JUL,AUG,SEP,OKT,NOV,DEC
4500 DATA 0,31,59,90,120,151,181,212,243,273,304,334


```

Muurstoter



Deze keer heeft u een erg ongewone doelstelling, want u moet uzelf laten neerstorten om punten te verzamelen.

Laten we hopen dat wat het Commando Hoofdkwartier ons heeft opgedragen, zin heeft.



Hoe te spelen

Er komen drie muren van verschillende constructie en kleur over het scherm rollen.

Elke muur heeft gaten in z'n structuur en u moet door de eerste twee heenschieten en de derde raken. Als u denkt dat het gemakkelijk is, dan kunnen we u vertellen dat de muren bewegen met verschillende snelheden ten opzichte van elkaar, en dat de gaten niet alleen van verschillende grootte zijn maar zelfs van muur tot muur verschillen.

U dient de muren van tevoren zien aankomen, de gaten ontdekken en uzelf lanceren door op de SPATIEBALK te drukken.

Gebruik, om te richten, de toetsen A en Z.

U heeft 10 schoten en u krijgt het maximum aantal punten als u de derde muur raakt.

Als u precies 150 punten heeft vergaard, krijgt u 10 bonuspunten en 10 extra levens.

Programma



```
10 REM MUURSTOTER MET M/CODE BEWEGING
20 GRAPHICS 5:POKE 752,1:PRINT
30 GOSUB 1000
40 GOSUB 1500
50 F=0:S=65
60 Y=100:X=S:GOSUB 3000
70 POKE H1,100:POKE H2,150:POKE H3,200
80 POKE C1,200:POKE C2,100:POKE C3,50
90 POKE C4,38:G=10:N=10:SC=0
100 GOSUB 2000
110 IF F=1 THEN 300
120 IF G=0 THEN 400
130 IF PEEK(764)=23 THEN Z=USR(12381)
140 IF PEEK(764)=63 THEN Z=USR(12412)
150 IF PEEK(764)=33 THEN F=1:G=G-1:POKE 764,0
200 GOTO 100
300 X=X+2:POKE 53278,1:POKE H4,X
310 IF PEEK(53263)=0 THEN 350
320 SC=SC+10*PEEK(53263)
330 PRINT " ";10*PEEK(53263):X=220
340 IF SC=150 THEN SC=SC+10:N=20:G=G+10:PRINT " 10 BO
NUS SCHOTEN"
350 IF X<210 THEN 100
360 F=0:X=S:POKE H4,X:GOTO 100
370 REM *****
400 GRAPHICS 1+16
410 POKE H1,0:POKE H2,0:POKE H3,0:POKE H4,0
420 POSITION 0,3:PRINT #6;" einde spel"
430 POSITION 0,10:PRINT #6:N;" SCHOTEN GEVUURD"
440 POSITION 0,17:PRINT #6;" SCORE :";SC
450 OPEN #1,4,0,"K:"
460 GET #1,A:IF A<>32 THEN 460
470 CLOSE #1
480 RUN
490 REM *****
1000 REM SPELERS OPZETTEN
1010 POKE 559,62:POKE 53277,3
1020 PMBASE=40:POKE 54279,PMBASE
1030 H1=53248:H2=H1+1:H3=H2+1:H4=H3+1
1040 C1=704:C2=C1+1:C3=C2+1:C4=C3+1
1050 POKE H1,0:POKE H2,0:POKE H3,0:POKE H4,0
1060 P1=PMBASE*256+1024:P2=P1+256:P3=P2+256:P4=P3+256
1070 FOR I=0 TO 245 STEP 2
1080 POKE P1+I,170:POKE P1+I+1,85
1090 POKE P2+I,170:POKE P2+I+1,85
1100 POKE P3+I,170:POKE P3+I+1,85
1110 POKE P4+I,0:POKE P4+I+1,0:NEXT I
```


```
1120 FOR I=0 TO 192 STEP 32:FOR L=0 TO 31
1130 POKE P2+L+I,0:NEXT L:NEXT I
1140 FOR I=0 TO 127:POKE P3+I,0:NEXT I
1150 FOR I=0 TO 224 STEP 32:FOR L=0 TO 15
1160 POKE P1+L+I,0:NEXT L:NEXT I
1170 RETURN
1500 REM INLEZEN MACHINE-CODE
1510 RESTORE 4000
1520 FOR L=12288 TO 12442
1530 READ D:POKE L,D
1540 NEXT L
1550 RETURN
2000 REM MUREN BEWEGEN
2010 Z=USR(12288):Z=USR(12311)
2020 Z=USR(12342):RETURN
3000 REM SCHIP TEKENEN
3010 POKE H4,X
3020 POKE P4+Y,240:POKE P4+Y+1,112:POKE P4+Y+2,60
3030 POKE P4+Y+3,112:POKE P4+Y+4,240
3040 RETURN
4000 DATA 173,255,44,72,162,254,189,0,44,157,1,44,202
,224,255,208,245
4010 DATA 104,141,0,44,104,96
4020 DATA 173,254,45,72,173,255,45,72,162,253,189,0,4
5,157,2,45,202,224,255
4030 DATA 208,245,104,141,1,45,104,141,0,45,104,96
4040 DATA 173,253,46,72,173,254,46,72,173,255,46,72,1
62,252
4050 DATA 189,0,46,157,3,46,202,224,255,208,245
4060 DATA 104,141,2,46,104,141,1,46,104,141,0,46,104,
96,173,254,47,72
4070 DATA 173,255,47,72,162,253,189,0,47,157,2,47,202
,224,255,208,245,104,141,1,47,104,141,0,47,104,96
4080 DATA 173,0,47,72,173,1,47,72,162,0,189,2,47,157,
0,47,232,224,253
4090 DATA 208,245,104,141,255,47,104,141,254,47,104,9
6
```

Voor de opkomst van raketten e.d. gingen onderzeeboten in groepen achter nietsvermoedende koopvaardijsschepen aan die geen weerstand konden bieden.

We hebben uit deze situatie spelmogelijkheden gehaald en de regels van het spel in wat leukere veranderd. Deze keer kunnen de onderzeeboten niet schieten.

Een groep vijandelijke onderzeeërs is onderweg naar uw kruiser. Ofschoon er veel onderzeeboten in uw gebied zijn, zullen er slechts twee tegelijk op het scherm verschijnen.

Een uit de groep is langzamer dan de andere en het moet gemakkelijk zijn om deze met uw dieptebommen te raken. De leider is echter snel en duikt meteen naar de zeebodem.



Hoe te spelen

De onderzeeboten beginnen bij de oppervlakte en proberen naar de bodem te duiken in een poging te ontsnappen. U dient ze tot zinken te brengen met uw dieptebommen voordat ze zo weggelopen. Als er drie zijn ontsnapt, eindigt het spel. Gebruik de Z en X voor links en rechts en de SPATIE om de dieptebommen te laten vallen.

U krijgt 30 punten voor de snelle leider en 15 voor de langzamere.

Programmeertips

Om de grootte van de onderzeeboten zelf te bepalen om ze gemakkelijker te kunnen raken, kunt u de waarde van V in regel 40 in 1 veranderen. Hiermee worden uw doelen tweemaal zo groot.

Om de snelheid van de dieptebommen te veranderen, kan in regel 200 een enkele maal een extra Z=USR(12381) worden toegevoegd of weggelaten.

Programma

```

10 REM ONDERZEEBOOTJAGER
20 GOSUB 460:GRAPHICS 5:POKE 752,1:PRINT
30 WV=1:E=0:GOTO 380
40 V=0:Y=80:GOSUB 490
50 X=100:POKE H1,X
60 A=40:B=200:POKE H2,A:POKE H3,B:C=1:D=-3
70 POKE 53256,V:POKE 53257,V:POKE 53258,V
80 P=0:Q=0:POKE 623,1
90 SETCOLOR 4,9,4:SETCOLOR 2,9,4
100 POKE C1,200:POKE C2,90:POKE C3,158
110 IF PEEK(764)=23 AND X>40 THEN X=X-2:POKE H1,X
120 IF PEEK(764)=22 AND X<200 THEN X=X+2:POKE H1,X
130 IF PEEK(764)=33 AND F=0 THEN F=1:POKE H4,X+4:GOSU
B 530
140 IF A=0 THEN 170
150 A=A+C:POKE H2,A:IF A<40 OR A>200 THEN C=-C:GOSUB
550
160 IF B=0 THEN 180
170 B=B+D:POKE H3,B:IF B<40 OR B>200 THEN D=-D:GOSUB
590
180 IF F=0 THEN 260
190 F=F+1:IF F=50 THEN F=0:POKE 764,0:GOTO 260
200 POKE 53278,1:Z=USR(12381):Z=USR(12381)
210 IF PEEK(53263)=0 THEN 260
220 SC=SC+15*PEEK(53263):PRINT "          ";SC
230 FOR L=0 TO 240 STEP 3:SOUND 2,L,4,14:NEXT L:SOUND
2,0,0,0
240 IF PEEK(53263)=2 THEN A=0:POKE H2,0:GOTO 260
250 B=0:POKE H3,0
260 IF A+B<>0 THEN 110
270 IF E<3 THEN WV=WV+1:GOTO 30
280 GRAPHICS 1:POSITION 0,5:PRINT #6;" ER ZIJN DRIE
ONTSNAPT"
290 POKE H1,0:POKE H2,0:POKE H3,0:POKE H4,0
300 POSITION 0,9:PRINT #6;" het spel is af
gelopen"
310 POSITION 0,12:PRINT #6;" SCORE : ";SC
320 POSITION 0,14:PRINT #6;"MOEILIJKHEIDSGRAAD:";WV
330 POSITION 0,17:PRINT #6;" SPATIE"

```

```

340 OPEN #1,4,0,"K:"
350 GET #1,A:IF A<>32 THEN 350
360 CLOSE #1
370 RUN
380 REM
390 POKE 559,62:POKE 53277,3
400 PMBASE=40:POKE 54279,PMBASE
410 H1=53248:H2=H1+1:H3=H2+1:H4=H3+1
420 C1=704:C2=C1+1:C3=C2+1:C4=C3+1
430 POKE H1,0:POKE H2,0:POKE H3,0:POKE H4,0
440 P1=PMBASE*256+1024:P2=P1+256:P3=P2+256:P4=P3+256:
Z=USR(12396)
450 PRINT CHR$(125);"OPDRACHT NUMMER ";WV;" ":GOTO 40
460 REM
470 RESTORE 630:FOR L=12288 TO 12416
480 READ D:POKE L,D:NEXT L:RETURN
490 REM SCHEPEN
500 RESTORE 740:FOR L=0 TO 7:READ A,B,C
510 POKE P1+L+48,A:POKE P2+L+Y,B
520 POKE P3+L+Y,C:NEXT L:RETURN
530 REM BODEM
540 POKE P4+60,24:POKE P4+61,60:POKE P4+62,60:POKE P4
+63,24:RETURN
550 REM NR 1 NEER
560 FOR L=1 TO 10:SOUND 1,10*L,10,14:Z=USR(12311):NEX
T L:SOUND 1,0,0,0
570 P=-1:IF P=8 THEN A=0:POKE H2,0:E=E+1
580 RETURN
590 REM NR 2 NEER
600 FOR L=1 TO 10:SOUND 1,20*L,10,14:Z=USR(12342):NEX
T L:SOUND 1,0,0,0
610 Q=Q+1:IF Q=6 THEN B=0:POKE H3,0:E=E+1
620 RETURN
630 DATA 173,255,44,72,162,254,189,0,44,157,1,44,202,
224,255,208,245
640 DATA 104,141,0,44,104,96
650 DATA 173,254,45,72,173,255,45,72,162,253,189,0,45
,157,2,45,202,224,255
660 DATA 208,245,104,141,1,45,104,141,0,45,104,96
670 DATA 173,253,46,72,173,254,46,72,173,255,46,72,16
2,252
680 DATA 189,0,46,157,3,46,202,224,255,208,245
690 DATA 104,141,2,46,104,141,1,46,104,141,0,46,104,9
6
700 DATA 162,253,189,0,47,157,2,47,202,224,255
710 DATA 208,245,104,96
720 DATA 162,0,169,0,157,0,44,157,0,45,157,0,46,157,0
,47
730 DATA 232,208,241,104,96
740 DATA 64,4,32

```




750 DATA 94,4,32
760 DATA 82,4,32
770 DATA 243,126,126
780 DATA 255,193,131
790 DATA 127,193,131
800 DATA 63,126,126
810 DATA 0,0,0

Brandkastkraker



Dit is uw kans om te zien of u ook tot de categorie MEESTERKRAKERS behoort. We hebben een paar codes gemaakt die u moet doorbreken om de brandkast te kunnen openen. Getallen van 1 tot 8 zijn mogelijk.



Hoe te spelen

De computer zal willekeurig een 4-cijferig getal in z'n geheugen opnemen en u vragen naar uw eerste antwoord.

Toets uw nummer in en druk op RETURN.

Na uw eerste antwoord zal uw computer zeggen hoeveel cijfers goed waren en hoeveel ook op de goede plaats stonden.

Er zal niet worden verteld welke cijfers op de goede plaats staan, alleen maar hoeveel. U dient er achter te komen welke cijfers dat zijn.

We zeiden dat het een test was voor 'meesterkrakers', nietwaar?

Programma



```
10 REM BRANDKASTKRAKER
20 DIM C(3),G(3),T(3)
30 GOSUB 1000
40 GRAPHICS 1:POKE 710,0
50 G=0:F=0:POKE 752,1
60 OPEN #1,4,0,"K:"
80 GOSUB 1500
90 REM *****
100 G=G+1:IF G=11 THEN 900
110 GOSUB 2000
120 GOSUB 3000
130 IF F=0 THEN 100
200 REM NUMMER GERADEN
210 POSITION 0,16:PRINT #6;"GOED ";
220 FOR L=0 TO 3
230 ? #6;C(L);" ";
240 NEXT L
250 POSITION 0,18:PRINT #6;"      IN ";G;" BEURTEN"
260 GOTO 940
900 REM NUMMER NIET GERADEN
910 POSITION 0,16:PRINT #6;"HET NUMMER WAS:"
920 POSITION 2,18
930 FOR L=0 TO 3:PRINT #6;C(L);" ";:NEXT L
940 GET #1,A:IF A<>32 THEN 940
950 CLOSE #1
960 RUN
1000 REM NUMMERS OPZETTEN
1010 C(0)=INT(RND(1)*8+1)
1020 FOR L=1 TO 3
1030 C(L)=INT(RND(1)*8+1)
1040 FOR I=0 TO L-1
1050 IF C(L)=C(I) THEN C(L)=INT(RND(1)*8+1):GOTO 1050
1060 NEXT I
1070 NEXT L
1080 RETURN
1500 REM TITEL KOLOMMEN
1510 POSITION 0,0:PRINT #6;"RAAD mijn NUMMER"
1520 ? "          GOED EN      NUMMERS"
1530 ? "          GOEDE PLAATS    GOED"
1540 RETURN
2000 REM RADEN
2010 L=0
2020 GET #1,D:D=D-48
2025 IF D=78 THEN 2100
2027 IF D=107 THEN 2200
2030 IF D<1 OR D>8 THEN 2020
2035 IF L=4 THEN 2020
```


```
2040 POSITION 2+L*2,G+2:PRINT #6;CHR$(D+144)
2050 G(L)=D:L=L+1:GOTO 2020
2100 L=L-1:IF L=-1 THEN L=0
2110 POSITION 2+L*2,G+2:PRINT #6;" ";
2120 GOTO 2020
2200 IF L<3 THEN 2020
2210 RETURN
3000 REM SCORE G(.) TEGEN C(.)
3010 B1=0:W1=0
3020 FOR H=0 TO 3
3030 IF G(H)=C(H) THEN B1=B1+1
3040 NEXT H
3050 FOR L=0 TO 3:T(L)=C(L):NEXT L
3060 FOR J=0 TO 3
3070 IF T(J)=-1 THEN 3110
3080 FOR L=0 TO 3
3090 IF T(J)=G(L) THEN W1=W1+1:T(J)=-1
3100 NEXT L
3110 NEXT J
3120 W1=W1-B1
3130 IF B1=4 THEN F=1
3200 POSITION 12,2+G
3210 PRINT #6;B1;" ";W1
3220 RETURN
```

Fruitmachine



Dit is een goedkope manier om te gokken zonder het risico te lopen iets te verliezen.

U krijgt een bepaald bedrag in het begin van het spel en u speelt tegen het noodlot.



Hoe te spelen

Uw ATARI computer zal een venster laten zien waar tekeningetjes in staan. De voorspellingen staan op schijven die de computer vervolgens laat draaien en op willekeurige momenten laat stoppen met een serie van drie tekeningen in het midden.

Bij twee gelijke afbeeldingen heeft u een kleine prijs en bij drie, vanzelfsprekend, een grotere. Er staan verschillende 'vasthoudmogelijkheden' tot uw beschikking en die kunt u kiezen door de cijfers 1, 2 of 3 in te toetsen. Mocht u zich bedenken dan kunt u alle schijven weer loslaten door op de R-toets te drukken.

Als de derde schijf op een A stopt, dan gaat de computer automatisch verder zoeken totdat er drie gelijke afbeeldingen voorkomen.

Druk, nadat u heeft gekozen welke schijf u wilt vasthouden, op de SPATIEBALK om de overgebleven schijven weer te draaien. Als u niets wilt of kunt houden, druk dan op RETURN voor de volgende ronde.

Het spel eindigt wanneer u zonder contanten achterblijft of als u op de BREAK-toets drukt.

Programmeertips

Om uw winstkans te veranderen, kunt u de DATA wijzigen vanaf regel 9000 tot regel 9150. Getallen tussen 0 en 9 zijn geoorloofd.

Wilt u de prijzen veranderen, verander dan de DATA in de regels 20 000 tot 20 010. Dit houdt wel in dat u zowel meer kunt winnen als meer kunt verliezen.

Programma

```

10 REM FRUIT MACHINE
11 REM MET M/CODE ROUTINE
15 DIM A(31),B(31),C(31),G(79)
17 DIM P(3),H(3)
20 GOSUB 1000:GOSUB 2000
30 GOSUB 3000:POKE 752,1
50 POKE H1,100:POKE H2,124:POKE H3,148
60 M=20
65 COLOR 3:PLOT 10,12:PLOT 29,12
70 PRINT "KREDIETEN :";
80 P(1)=16:P(2)=16:P(3)=16
90 POKE C1,88:POKE C2,200:POKE C3,38
100 PRINT STR$(M);"   ":PRINT
110 IF M=0 THEN 900
115 H(1)=0:H(2)=0:H(3)=0
117 COLOR 2:PLOT 13,8:DRAWTO 26,8
120 IF RND(1)>0.7 THEN GOSUB 4000
200 REM SCHIJVEN DRAAIEN
202 M=M-1
205 IF H(1)+H(2)+H(3)=3 THEN 400
210 GOSUB 4100
220 IF RND(1)<0.95 THEN 210
230 H(1)=1
240 GOSUB 4100
250 IF RND(1)<0.92 THEN 240
260 H(2)=1
270 GOSUB 4100
280 IF RND(1)<0.9 THEN 270
290 REM SCHIJVEN ALLEMAAL GESTOPT
400 REM SCORE BEREKENEN
410 P=A(P(1)):Q=B(P(2)):R=C(P(3))
420 IF R<>9 THEN 450
425 PRINT "      AUTOMATISCHE LOOP   ":PRINT
427 FOR I=200 TO 0 STEP -3:SOUND 0,I,10,14:NEXT I
428 SOUND 0,0,0,0
430 H(1)=1:H(2)=0:H(3)=0
432 GOSUB 4100:IF P<>B(P(2)) THEN 432
433 H(2)=1
434 GOSUB 4100:IF P<>C(P(3)) THEN 434
440 GOTO 410

```



```
450 IF P<>Q THEN 540
460 IF Q=R THEN 500
469 REM PAAR LINKS
470 RESTORE 20000:REM SCORES
480 FOR I=1 TO P+1:READ D:NEXT I
490 M=M+D:PRINT "    PAAR SCOORT :";STR$(D):PRINT :GOT
O 550
499 REM ALLEMAAL HETZELFDE
500 RESTORE 20010
510 FOR I=1 TO P+1:READ D:NEXT I
520 M=M+D:PRINT "    DRIE SCOORT : ";STR$(D):PRINT :GOT
O 550
530 REM GEEN SCORES
540 PRINT "    GEEN WINNENDE SERIE":PRINT
549 REM EINDE VAN SCORE
550 OPEN #1,4,0,"K:"
560 GET #1,A
570 IF A<>155 THEN 560
580 CLOSE #1
585 PRINT "                                ":PRINT ""
590 GOTO 100
900 REM EINDE VAN KREDIET
910 PRINT "    *** BLUT ***    "
920 PRINT "<SPATIE> OM OVERNIEUW TE BEGINNEN"
930 OPEN #1,4,0,"K:"
940 GET #1,A
950 IF A<>32 THEN 940
960 CLOSE #1
970 RUN
1000 REM SPEELVELD TEKENEN
1010 GRAPHICS 3
1012 SETCOLOR 2,3,4
1015 COLOR 2:SETCOLOR 1,9,2
1020 FOR L=0 TO 23:PLOT 0,L
1030 DRAWTO 39,L:NEXT L
1040 COLOR 4
1050 FOR L=10 TO 14:PLOT 10,L
1060 DRAWTO 29,L:NEXT L
2000 REM MACHINE CODE INLEZEN
2005 RESTORE 10000
2010 FOR I=12288 TO 12380
2020 READ D:POKE I,D
2030 NEXT I
2040 RETURN
3000 REM SPELERS INITIALISEREN
3010 PMBASE=40:POKE 54279,PMBASE
3015 POKE 559,62:POKE 53277,3
3020 P1=PMBASE*256+1024
3030 P2=P1+256:P3=P2+256
3040 C1=704:C2=C1+1:C3=C2+1
```

```
3050 H1=53248:H2=H1+1:H3=H2+1
3060 POKE H1,0:POKE H2,0:POKE H3,0
3070 RESTORE 9500
3080 POKE 623,4
3100 FOR L=0 TO 79
3110 READ D:G(L)=D
3120 NEXT L
3125 RESTORE 9000
3130 FOR L=0 TO 31
3140 READ P,Q,R
3150 A(L)=P:B(L)=Q:C(L)=R
3155 S=L*8
3160 FOR I=0 TO 7
3170 POKE P1+S+I,G(P*8+I)
3180 POKE P2+S+I,G(Q*8+I)
3190 POKE P3+S+I,G(R*8+I)
3200 NEXT I
3210 NEXT L
3220 RETURN
4000 REM HOUD ROUTINE
4005 H(1)=0:H(2)=0:H(3)=0
4007 COLOR 2:PLOT 13,8:DRAWTO 26,8
4010 OPEN #1,4,0,"K:"
4020 PRINT "      WELKE HOUDEN"
4030 PRINT
4040 GET #1,K
4050 IF K=32 THEN PRINT "              ":PRINT
      :CLOSE #1:RETURN
4060 IF K=ASC("R") THEN CLOSE #1:GOTO 4005
4070 IF K<49 OR K>51 THEN 4040
4075 COLOR 3:PLOT 7+6*(K-48),8:PLOT 8+6*(K-48),8
4080 H(K-48)=1:GOTO 4040
4100 REM SCHIJVEN LATEN DRAAIEN
4110 FOR Q=0 TO 3
4120 IF H(1)=0 THEN A=USR(12288)
4130 IF H(2)=0 THEN A=USR(12319)
4140 IF H(3)=0 THEN A=USR(12350)
4150 NEXT Q
4160 FOR Q=1 TO 3
4170 IF H(Q)<>1 THEN P(Q)=P(Q)-1
4180 IF P(Q)=-1 THEN P(Q)=31
4190 NEXT Q:RETURN
8999 REM SCHIJF DATA
9000 DATA 0,1,4,1,3,3
9010 DATA 2,5,1,3,2,4
9020 DATA 4,4,9,5,6,6
9030 DATA 6,7,5,7,8,8
9040 DATA 8,6,4,9,3,3
9050 DATA 8,2,1,7,4,4
9060 DATA 6,7,2,5,9,4
```

```

9070 DATA 4,1,1,3,2,8
9080 DATA 2,3,2,1,6,9
9090 DATA 2,7,4,3,5,1
9100 DATA 4,4,2,5,4,1
9110 DATA 6,3,1,7,8,4
9120 DATA 8,1,6,7,7,2
9130 DATA 6,3,7,5,6,5
9140 DATA 4,2,1,3,4,4
9150 DATA 2,1,2,1,5,3
9499 REM VORM DATA
9500 DATA 0,24,36,211,203,36,24,0
9510 DATA 8,16,86,127,255,126,126,60
9520 DATA 0,255,85,85,85,85,255,0
9530 DATA 0,63,127,124,248,252,126,63
9540 DATA 129,66,36,24,24,36,66,129
9550 DATA 56,186,146,124,16,40,68,68
9560 DATA 0,7,3,5,8,16,32,64
9570 DATA 24,24,24,255,255,24,24,24
9580 DATA 0,66,231,255,126,60,60,24
9590 DATA 0,24,36,36,126,66,66,66
9999 DATA MACHINE CODE DATA
10000 DATA 173,254,44,72,173,255,44,72
10010 DATA 162,253,189,0,44,157,2,44,202,224,255
10020 DATA 208,245,104,141,1,44
10030 DATA 104,141,0,44,104,96
10050 DATA 173,254,45,72,173,255,45,72
10060 DATA 162,253,189,0,45,157,2,45,202,224,255
10070 DATA 208,245,104,141,1,45
10080 DATA 104,141,0,45,104,96
10100 DATA 173,254,46,72,173,255,46,72
10110 DATA 162,253,189,0,46,157,2,46
10120 DATA 202,224,255,208,245,104,141,1,46
10130 DATA 104,141,0,46,104,96
20000 DATA 1,1,2,2,3,3,4,4,5,5
20010 DATA 2,3,4,5,6,7,8,9,10,20


```

Flitsende graphics

Omdat de ATARI XL computer een grote keuze aan kleur- en graphics-mogelijkheden biedt, dachten we eraan deze te laten zien aan u en uw vrienden. Er zijn drie demonstraties waar u uit kunt kiezen vanuit een 'menu'.

- Het eerste programma zal de kleuren en tinten laten zien.
- Het tweede programma zal een ingewikkeld mozaïekeffect geven.
- Het derde programma zal een bal laten zien die een kometenstaart achter zich aan heeft.

Alle programma's zijn ideaal voor demonstraties en kunnen worden gebruikt als aandachtstrekker in een etalage.



Hoe te gebruiken

Gebruik de 1, 2 of 3 om te kiezen. Gebruik de SPATIEBALK om naar het menu terug te keren.

Programma

```

10 REM FLITSENDE GRAPHICS
20 DIM A$(2)
30 GRAPHICS 1+16:POSITION 0,4
40 PRINT #6;"    UW DEMO NUMMER"
50 POSITION 4,7:PRINT #6;"( 1 2 OF 3 )"
60 OPEN #1,4,0,"K:"
70 GET #1,A:A$=CHR$(A)
80 IF A$<"1" OR A$>"3" THEN 70
90 CLOSE #1
100 ON A-48 GOSUB 1000,2000,3000
110 GOTO 30
120 REM
990 FOR T=1 TO 200:NEXT T:RETURN
1000 REM DEMO EEN
1010 GRAPHICS 9+16
1020 SETCOLOR 4,0,0
1030 FOR X=0 TO 15
1040 COLOR X
1050 FOR D=0 TO 2
1060 C=X*5+D:PLOT C,0:DRAWTO C,191
1070 NEXT D:NEXT X
1100 FOR C=0 TO 15
1110 SETCOLOR 4,C,0:GOSUB 990:NEXT C
1120 SETCOLOR 4,0,0
1130 GRAPHICS 11+32:REM NAAR 16 KLEUREN
1140 FOR C=0 TO 15
1150 SETCOLOR 4,0,C:GOSUB 990:NEXT C
1160 RETURN
2000 REM DEMO TWEË
2010 GRAPHICS 15+16
2015 SETCOLOR 0,2,14:SETCOLOR 1,12,14
2020 FOR X=0 TO 156 STEP 4
2030 COLOR 1:PLOT X,0:DRAWTO 156,X
2040 PLOT 156-X,156:DRAWTO 0,156-X
2050 COLOR 2:PLOT 156,X:DRAWTO 156-X,156
2060 PLOT 0,156-X:DRAWTO X,0
2070 NEXT X
2080 SETCOLOR 2,0,14
2090 COLOR 3
2100 FOR X=0 TO 32
2110 PLOT 78+X,78:DRAWTO 78,78+X:DRAWTO 78-X,78
2120 DRAWTO 78,78-X:DRAWTO 78+X,78
2130 NEXT X
2140 FOR C=14 TO 0 STEP -2
2150 SETCOLOR 0,2,C:SETCOLOR 1,12,14-C
2160 SETCOLOR 2,0,C:NEXT C
2170 IF PEEK(764)=33 THEN 2220

```

```

2180 FOR C=0 TO 14 STEP 2
2190 SETCOLOR 0,2,C:SETCOLOR 1,12,14-C:SETCOLOR 2,0,C
2200 FOR T=1 TO 400:NEXT T
2210 NEXT C
2220 RETURN
3000 REM DEMO DRIE
3010 GRAPHICS 15+16
3020 SETCOLOR 2,2,14
3030 SETCOLOR 1,2,10:SETCOLOR 0,2,6
3040 X=0:Y=150
3050 A=2.5:B=6
3060 E=0:F=0:G=0:H=0
3070 K=0:L=0:M=0:N=0:O=0:P=0
3100 C=X+A:D=Y+B
3110 IF C<0 OR C>159 THEN A=-A:GOTO 3100
3120 IF D<0 OR D>190 THEN B=-0.95*B:GOTO 3100
3200 COLOR 4:PLOT INT(O),INT(P)
3210 COLOR 1:PLOT INT(M),INT(N)
3220 PLOT INT(K),INT(L)
3230 COLOR 2:PLOT INT(I),INT(J)
3240 PLOT INT(G),INT(H)
3250 COLOR 3:PLOT INT(E),INT(F)
3260 PLOT INT(C),INT(D)
3270 O=M:P=M:M=K:N=L
3280 K=I:L=J:I=G:J=H
3290 G=E:H=F:E=X:F=Y
3300 X=C:Y=D
3310 B=B+0.11
3320 IF PEEK(764)=33 THEN RETURN
3330 GOTO 3100

```


Akkoordenorgel



Neem nu de kans om die musikale talenten te laten horen waar u al jaren trots op was.

Dit programma maakt van uw ATARI XL computer een mini elektronisch-orgeltje en dat voor de prijs van een paar minuten typen.

In plaats van slechts een paar toetsen aan te wijzen om in te drukken, hebben wij een paar akkoorden ingeprogrammeerd.



Hoe te spelen

De cijfers geven de akkoorden weer. Druk een cijfer om te kiezen.

De toetsen in de rij van Q en A geven de melodie weer.

Als u een pauze wilt inlassen en als u niet continu de laatst gekozen noot wilt horen, druk dan op RETURN. Als u echter een solo noten-riedeltje weg wilt geven tijdens het spelen van de akkoorden, druk dan op de SPATIE-balk.

We wensen u en de ontroostbare familieleden vele leuke uren speelgenot.

Programma



```
10 REM AKKOORDEN--ORGEL
100 K=PEEK(764)
110 IF K=31 THEN SOUND 1,251,10,10:SOUND 2,194,10,10:
SOUND 3,164,10,10
120 IF K=30 THEN SOUND 1,162,10,10:SOUND 2,128,10,10:
SOUND 3,108,10,10
130 IF K=26 THEN SOUND 1,217,10,10:SOUND 2,173,10,10:
SOUND 3,144,10,10
140 IF K=24 THEN SOUND 1,144,10,10:SOUND 2,114,10,10:
SOUND 3,96,10,10
150 IF K=29 THEN SOUND 1,193,10,10:SOUND 2,153,10,10:
SOUND 3,128,10,10
160 IF K=27 THEN SOUND 1,128,10,10:SOUND 2,102,10,10:
SOUND 3,85,10,10
170 IF K=51 THEN SOUND 1,187,10,10:SOUND 2,148,10,10:
SOUND 3,124,10,10
180 IF K=53 THEN SOUND 1,251,10,10:SOUND 2,204,10,10:
SOUND 3,164,10,10
190 IF K=48 THEN SOUND 1,162,10,10:SOUND 2,136,10,10:
SOUND 3,108,10,10
200 IF K=50 THEN SOUND 1,217,10,10:SOUND 2,184,10,10:
SOUND 3,144,10,10
210 IF K=33 THEN SOUND 0,0,0,0:SOUND 1,0,0,0:SOUND 2,
0,0,0,0:SOUND 3,0,0,0
220 IF K=12 THEN SOUND 0,0,0,0
230 IF K=63 THEN SOUND 0,124,10,14
240 IF K=46 THEN SOUND 0,114,10,14
250 IF K=62 THEN SOUND 0,108,10,14
260 IF K=42 THEN SOUND 0,102,10,14
270 IF K=58 THEN SOUND 0,96,10,14
280 IF K=56 THEN SOUND 0,91,10,14
290 IF K=45 THEN SOUND 0,85,10,14
300 IF K=61 THEN SOUND 0,81,10,14
310 IF K=43 THEN SOUND 0,76,10,14
320 IF K=57 THEN SOUND 0,72,10,14
330 IF K=11 THEN SOUND 0,68,10,14
340 IF K=1 THEN SOUND 0,64,10,14
350 IF K=5 THEN SOUND 0,61,10,14
360 IF K=8 THEN SOUND 0,57,10,14
370 IF K=0 THEN SOUND 0,53,10,14
380 IF K=10 THEN SOUND 0,50,10,14
390 IF K=2 THEN SOUND 0,47,10,14
400 IF K=6 THEN SOUND 0,45,10,14
410 IF K=15 THEN SOUND 0,42,10,14
420 IF K=7 THEN SOUND 0,40,10,14
430 GOTO 100
```

Robot-eiland



Het wordt een onsmakelijk experiment voor degene die niet van robots, geesten, water en drijfzand houden.

U bent de enige die aan het plan van de gekke professor om mensen in robots te veranderen is ontkomen en u moet langs een eiland met robots en geesten zien te komen.

U kunt de robots vernietigen (als vernietigen het juiste woord is) door ze naar het drijfzand te lokken of naar de rotsen waar ze misschien struikelen en in zee zullen vallen.

U kunt de geesten niet doden (omdat ze al dood zijn) maar u kunt wel voorkomen dat ze u vangen of dwarszitten door een robot te vernietigen. Dit ontslaat de geest van zijn plichten.



Hoe te spelen

Als het spel begint, zal de geest links verschijnen en de robot rechtsboven. Het drijfzand is links boven. Onthoud dat u ook het slachtoffer kunt worden van drijfzand, dus ga niet te ver naar die hoek.

Gebruik de Z en X voor links en rechts en de / en + voor op en neer.

Uw score verschijnt aan het eind van het spel en u kunt overnieuw beginnen door op de SPATIEBALK te drukken.

Programmeertips

Verander de punten voor elk karakter door de regels 50, 60 en 70 te wijzigen. Ook kunt u de grootte van het drijfzandgebied veranderen door de DATA in regel 1110 te wijzigen.

Om het spel gemakkelijker te maken voor jongere spelers, kan een nieuwe regel worden toegevoegd, namelijk:

```
105 FOR Z = 1 TO 40 : NEXT Z
```

Programma

```
10 REM ROBOT EILAND
20 GRAPHICS 5+16:POKE 623,4
30 GOSUB 1000:GOSUB 2000:GOSUB 3000
40 X1=120:Y1=75
50 X2=80:Y2=80
60 X3=140:Y3=50
70 POKE C1,13*16+12:POKE C2,4*16+8:POKE C3,6*16+10
80 POKE H1,X1:POKE H2,X2:POKE H3,X3
90 M=22:R1=1:R2=1:SC=0:T=500
100 K=PEEK(764):IF K=22 OR K=23 OR K=38 OR K=6 THEN M
    =K
110 IF M=22 THEN X1=X1+2:POKE H1,X1
120 IF M=23 THEN X1=X1-2:POKE H1,X1
130 IF M=38 THEN Y1=Y1+2:A=USR(12288)
140 IF M=6 THEN Y1=Y1-2:A=USR(12301)
200 POKE 53278,1
210 IF PEEK(53260)<>0 OR PEEK(53252)<>0 THEN 4000
300 REM EEN ROBOT BEWEGEN
310 POKE 53278,1
320 IF R1=0 THEN 500
330 IF R2=0 THEN 400
340 IF RND(1)<0.5 THEN 500
400 REM BEWEEG DE GEEST ROBOT
410 IF X2>X1 THEN X2=X2-2:POKE H2,X2
420 IF X2<X1 THEN X2=X2+3:POKE H2,X2
430 IF Y2>Y1 THEN Y2=Y2-2:A=USR(12327)
440 IF Y2<Y1 THEN Y2=Y2+2:A=USR(12314)
450 GOTO 600
500 REM TWEDE ROBOT
510 IF X3>X1 THEN X3=X3-2:POKE H3,X3
520 IF X3<X1 THEN X3=X3+2:POKE H3,X3
530 IF Y3>Y1 THEN Y3=Y3-2:A=USR(12353)
540 IF Y3<Y1 THEN Y3=Y3+2:A=USR(12340)
600 REM KIJK WAT ER GEBEURDE
650 IF R2=1 AND PEEK(53254)<>0 THEN GOSUB 5500
660 IF PEEK(53260)<>0 THEN 4000
670 IF T>0 THEN T=T-5
680 SC=SC+1
```

```

690 IF R2=1 THEN 100
700 REM BEIDE ROBOTS ZIJN DOOD
710 GOTO 4400
1000 REM EILAND TEKENEN
1010 COLOR 3:PLOT 5,0
1020 DRAWTO 50,3:DRAWTO 73,12
1030 DRAWTO 52,19:DRAWTO 76,34
1040 DRAWTO 36,47:DRAWTO 15,38
1050 DRAWTO 5,0
1100 COLOR 2
1110 PLOT 20,18:PLOT 20,18:PLOT 20,17
1120 RETURN
2000 REM SPELERS OPZETTEN
2010 PMBASE=40:POKE 54279,PMBASE
2020 POKE 559,62:POKE 53277,3
2030 H1=53248:H2=H1+1:H3=H2+1
2040 POKE H1,0:POKE H2,0:POKE H3,0
2050 C1=704:C2=C1+1:C3=C2+1
2060 P1=PMBASE*256+1024
2070 P2=P1+256:P3=P2+256
2080 FOR I=0 TO 255
2090 POKE P1+I,0:POKE P2+I,0:POKE P3+I,0
2100 NEXT I
2110 RESTORE 9000
2120 FOR I=0 TO 7
2130 READ A,B,C
2140 POKE P2+I+80,C:POKE P3+I+50,A
2150 POKE P1+I+75,B
2160 NEXT I
2170 RETURN
3000 REM MACHINE CODE INLEZEN
3010 RESTORE 8000
3020 FOR L=12288 TO 12365
3030 READ D
3040 POKE L,D
3050 NEXT L
3060 RETURN
4000 REM ALS DE SPELER VERLIEST
4010 FOR L=0 TO 240 STEP 3:POKE C1,L
4020 SOUND 0,L,10,10:SOUND 1,L+4,10,10
4030 SOUND 2,L+8,10,10:SOUND 3,L+12,10,10
4040 NEXT L
4050 FOR L=0 TO 3:SOUND L,0,0,0:NEXT L
4100 POKE H1,0:POKE H2,0:POKE H3,0
4110 GRAPHICS 1+16:POSITION 0,10
4120 PRINT #6;"gepakt door de robot"
4130 POSITION 4,14
4140 PRINT #6;"SCORE : ";STR$(SC)
4150 POSITION 6,17
4160 PRINT #6;"<spatie>"

```

```

4170 OPEN #1,4,0,"K:"
4200 GET #1,A
4210 IF A<>ASC(" ") THEN 4200
4220 CLOSE #1
4230 RUN
4400 REM ALS DE SPELER WINT
4410 POKE H1,0:POKE H2,0:POKE H3,0
4420 GRAPHICS 1+16:POSITION 2,10
4430 PRINT #6;"U HEBT DE ROBOT"
4440 GOTO 4130
5000 REM ROBOT 1 GEDOOD
5010 R1=0
5020 SC=SC+T
5030 POKE H2,0
5040 RETURN
5500 REM ROBOT 2 GEDOOD
5510 R2=0
5520 SC=SC+T
5530 POKE H3,0
5540 RETURN
7999 REM MACHINE CODE
8000 DATA 162,253,189,0,44,157,2,44,202,208,247,104,9
6
8010 DATA 162,0,189,2,44,157,0,44,232,208,247,104,96
8020 DATA 162,253,189,0,45,157,2,45,202,208,247,104,9
6
8030 DATA 162,0,189,2,45,157,0,45,232,208,247,104,96
8032 DATA 162,253,189,0,46,157,2,46,202,208,247,104,9
6
8050 DATA 162,0,189,2,46,157,0,46,232,208,247,104,96
8999 REM DE SPELERS VORMEN
9000 DATA 0,56,0
9001 DATA 0,56,0
9002 DATA 6,56,14
9003 DATA 15,150,15
9004 DATA 6,120,29
9005 DATA 9,16,25
9006 DATA 0,40,31
9007 DATA 0,68,62


```

Wereldvlaggen



Hoe dikwijls heeft u zich afgevraagd bij het voetballen, atletiek of de Olympische Spelen welke vlag van welk land is:

Uw zorgen zijn over na dit programma te hebben gebruikt.



Hoe te gebruiken

Uw computer zal, een paar seconden, de namen van drie landen laten zien.

U zult daarna, in mooie kleuren natuurlijk, een vlag van één van die landen zien.

Raad welk land het is door 1, 2 of 3 in te drukken.

Als uw antwoord fout is, zal het goede antwoord worden getoond. U zult na tien vragen een score krijgen.

Programmeertips

Als u een goede programmeur bent en u wilt experimenteren met de inhoud van het programma, dan kunt u in de regel 9000 en verder de kleuren, maten en vormen van de vlaggen veranderen.

Programma

```

10 REM WERELDVLAGGEN
20 OPEN #1,4,0,"K:"
30 DIM P(3)
40 DIM T$(20)
50 DIM A$(20)
60 SC=0:J=0
100 J=J+1:IF J=11 THEN 800
110 GOSUB 4000
120 GOSUB 5000
130 RESTORE 8900+G*100
140 GOSUB 1000
150 GET #1,D:IF D<ASC("1") OR D>ASC("3") THEN 150
200 IF P(D-48)=G THEN 400
300 REM FOUT ANTWOORD
310 GRAPHICS 1+16:POSITION 0,12
320 PRINT #6;"      FOUT"
330 FOR L=1 TO 200:NEXT L
350 POSITION 5,17:PRINT #6;A$
360 FOR L=1 TO 500:NEXT L
370 GOTO 100
400 REM GOED ANTWOORD
410 GRAPHICS 1+16:POSITION 0,12
420 PRINT #6;" JUIST : ";A$
430 FOR L=1 TO 500:NEXT L
440 SC=SC+1
450 GOTO 100
800 REM TIEN VRAGEN GEHAD
810 GRAPHICS 1+16
820 POSITION 0,8:PRINT #6;" EINDE VAN DE QUIZ"
830 POSITION 0,14:PRINT #6;"      U SCOORDE:";SC
840 POSITION 0,17:PRINT #6;"      VAN DE 10 BEURTEN"
850 POSITION 0,20:PRINT #6;"      < SPATIE >"
860 GET #1,A
870 IF A<>32 THEN 860
880 RUN
1000 REM VLAG TEKENEN
1010 GRAPHICS 7+16:SETCOLOR 4,0,6
1013 FOR L=0 TO 2:READ A,B:SETCOLOR L,A,B:NEXT L
1020 FOR L=0 TO 2:READ A,B:SETCOLOR L,A,B:NEXT L
1030 C=0
1040 C=C+1:COLOR C:IF C=4 THEN 1130
1050 READ X:IF X=0 THEN 1040
1060 READ Y,X1,Y1,CX,CY,N
1070 FOR L=1 TO N
1080 PLOT X,Y:DRAWTO X1,Y1
1090 X=X+CX:Y=Y+CY
1100 X1=X1+CX:Y1=Y1+CY

```



```
1110 NEXT L
1120 GOTO 1050
1130 RETURN
4000 REM 1 VLAG KIEZEN & 2 FOUTIEVE
4010 G=INT(RND(1)*8+1)
4020 W1=INT(RND(1)*8+1)
4030 IF W1=G THEN 4010
4040 W2=INT(RND(1)*8+1)
4050 IF W2=W1 OR W2=G THEN 4040
4060 RESTORE 10000
4070 FOR L=1 TO G:READ A$
4080 NEXT L
4090 RETURN
5000 REM DE DRIE KEUZES AFDRUKKEN
5010 GRAPHICS 1+16
5020 POSITION 0,4:PRINT #6;" DE KEUZES ZIJN:"
5030 POSITION 0,10:PRINT #6;" 1) ";
5040 POSITION 0,13:PRINT #6;" 2) ";
5050 POSITION 0,16:PRINT #6;" 3) ";
5060 P(1)=G:P(2)=W1:P(3)=W2
5070 FOR L=1 TO 10:T=INT(RND(1)*3+1)
5080 S=INT(RND(1)*3+1)
5090 TT=P(T):P(T)=P(S):P(S)=TT
5100 NEXT L
5110 FOR L=1 TO 3
5120 RESTORE 10000
5130 FOR I=1 TO P(L)
5140 READ T$:NEXT I
5150 POSITION 4,7+L*3
5160 PRINT #6;T$
5170 NEXT L
5180 FOR L=1 TO 700:NEXT L
5190 RETURN
8999 REM ** ZWEDEN **
9000 DATA 8,2,1,10,0,0
9010 DATA 1,1,1,38,1,0,50
9020 DATA 1,58,1,95,1,0,50
9030 DATA 70,1,70,38,1,0,90
9040 DATA 70,58,70,95,1,0,90
9050 DATA 0,1,39,1,57,1,0,159
9060 DATA 51,1,51,95,1,0,19
9070 DATA 0,0
9099 REM ** ITALIE **
9100 DATA 11,2,0,14,3,4
9110 DATA 1,1,1,95,1,0,52
9120 DATA 0,52,1,52,95,1,0,52
9130 DATA 0,103,1,103,95,1,0,52
9140 DATA 0,0
9199 REM ** FRANKRIJK **
9200 DATA 5,2,0,14,3,4
```

```


9210 DATA 1,1,1,95,1,0,52
9220 DATA 0,52,1,52,95,1,0,52
9230 DATA 0,103,1,103,95,1,0,52
9240 DATA 0,0
9299 REM ** BELGIE **
9300 DATA 0,0,1,10,3,2
9310 DATA 1,1,1,95,1,0,52
9320 DATA 0,52,1,52,95,1,0,52
9330 DATA 0,103,1,103,95,1,0,52
9340 DATA 0,0
9399 REM ** GRIEKENLAND **
9400 DATA 8,2,0,14,0,0
9410 DATA 1,1,1,20,1,0,25
9420 DATA 1,30,1,46,1,0,25
9422 DATA 40,1,40,20,1,0,25
9424 DATA 40,30,40,46,1,0,25
9426 DATA 65,1,159,1,0,1,10
9428 DATA 65,38,159,38,0,1,9
9430 DATA 65,21,159,21,0,1,9
9432 DATA 1,58,159,58,0,1,14
9434 DATA 1,82,159,82,0,1,12
9440 DATA 0,65,11,159,11,0,1,10
9450 DATA 65,30,159,30,0,1,9
9460 DATA 1,47,159,47,0,1,12
9465 DATA 1,70,159,70,0,1,12
9470 DATA 26,1,26,47,1,0,14
9480 DATA 1,21,64,21,0,1,9
9490 DATA 0,0
9499 REM ** NEDERLAND **
9500 DATA 3,2,0,14,8,2
9510 DATA 1,1,159,1,0,1,32
9520 DATA 0,1,32,159,32,0,1,32
9530 DATA 0,1,63,159,63,0,1,32
9540 DATA 0,0
9599 REM ** OOSTENRIJK **
9600 DATA 3,2,0,14,3,2
9610 DATA 1,1,159,1,0,1,32
9620 DATA 0,1,32,159,32,0,1,32
9630 DATA 0,1,63,159,63,0,1,32
9640 DATA 0,0
9699 REM ** DUITSLAND **
9700 DATA 0,0,3,2,1,10
9710 DATA 1,1,159,1,0,1,32
9720 DATA 0,1,32,159,32,0,1,32
9730 DATA 0,1,63,159,63,0,1,32
9740 DATA 0,0
9799 REM ** ALLE LANDEN NAMEN **
10000 DATA ZWEDEN,ITALIE,FRANKRIJK
10010 DATA BELGIE,GRIEKENLAND,NEDERLAND
10020 DATA OOSTENRIJK,DUITSLAND

```

Kruisraket-aanval



U heeft het commando over een lanceerinstallatie voor kruisraketten die tot taak heeft stationaire doelen aan de andere kant van de heuvelrug te vernietigen.



Hoe te spelen

U dient zelf de hoek en de snelheid van de lancering te bepalen, en te vuren met gebruik van de RETURN-toets.

De computer zal de baan van de raket automatisch bepalen waar zwaartekracht en luchtweerstand aan te pas komen.

Na elk schot zal de vijand zich verplaatsen en zullen de heuvels van vorm veranderen. U zult de baan van het projectiel na elk schot moeten herberekenen.

U heeft vijf raketten en elke voltreffer is tien punten waard.

Programmeertips

Om de weerstand van de atmosfeer te wijzigen, en daarbij de baan van de raket, kan regel 50 worden aangepast.

Ook kunt u de volgende regels toevoegen om de baan van het projectiel beïnvloedbaar te maken door middel van de Z- en X-toetsen:

```
182 IF PEEK(764) = 22 THEN C = C + 2
```

```
184 IF PEEK(764) = 23 THEN C = C - 2
```

```
186 POKE 764, 255
```

Programma

```

10 REM KRUISAANVAL
20 T=5
30 OPEN #1,4,0,"K:"
40 DEG
50 G=9.81
60 SC=0
70 GOSUB 510
80 GOSUB 640
90 GOSUB 730
100 REM KOGEL BEWEGEN
110 X=V*COS(TH);Y=V*SIN(TH)
120 A=10;B=10
130 C=A+X/30;D=B+Y/30
140 Y=Y-G/30;X=X-0.02
150 Z=0
160 IF C>159 THEN 380
170 IF B>79 THEN 200
180 IF D<=0 THEN 250
190 COLOR 4:PLOT A,79-B
200 IF D>79 THEN 230
210 LOCATE C,79-D,Z
220 COLOR 1:PLOT C,79-D
230 A=C;B=D
240 IF Z=0 THEN 130
250 REM KIJK WAT ER GEBEURT
260 IF Z=0 THEN 380
270 COLOR 1:PLOT C-3,80-D;DRAWTO C,79-D
280 DRAWTO C-2,77-D;PLOT C,79-D
290 DRAWTO C-1,82-D
300 ON Z-1 GOTO 310,380
310 REM DOEL GERAAKT
320 PRINT CHR$(125);"                DIE WAS RAAK"
330 FOR L=1 TO 200:SOUND 0,L,8,L/14:NEXT L
340 SC=SC+10
350 PRINT "                SCORE:";SC
360 SOUND 0,0,0,0
370 GOTO 430
380 REM DOEL GEMIST
390 PRINT CHR$(125);"                GEMIST                "
400 FOR L=0 TO 250:SOUND 0,L,10,14
410 SOUND 1,L+7,10,14:NEXT L
420 SOUND 1,0,0,0:SOUND 0,0,0,0
430 T=T-1:IF T<>0 THEN 100
440 GRAPHICS 1+16
450 POSITION 0,8:PRINT #6;" einde van opdracht"
460 POSITION 0,10:PRINT #6;" score :";SC
470 POSITION 0,12:PRINT #6;"                <SPATIE>"

```

```


480 GET #1,D:IF D<>32 THEN 480
490 CLOSE #1
500 RUN
510 REM HEUVEL TEKENEN
520 GRAPHICS 7
530 Y=79
540 COLOR 3:PLOT 19,79
550 FOR X=21 TO 93 STEP 3
560 DRAWTO X,Y
570 IF X<50 THEN Y=Y-INT(RND(1)*14)
580 IF X>55 THEN Y=Y+INT(RND(1)*11)
590 IF Y>79 THEN Y=79-INT(RND(1)*10)
600 IF Y<10 THEN Y=10+INT(RND(1)*5)
610 NEXT X
620 DRAWTO 95,79
630 RETURN
640 REM KANON+DOEL TEKENEN
650 COLOR 2:PLOT 0,79:DRAWTO 0,76:DRAWTO 4,76:DRAWTO
4,79
660 DRAWTO 0,79:DRAWTO 0,77:DRAWTO 7,74:DRAWTO 7,73
670 DRAWTO 4,76
680 X=INT(RND(1)*50+100)
690 PLOT X,79:DRAWTO X+8,75
700 PLOT X+8,79:DRAWTO X,75
710 PLOT X,79:DRAWTO X+8,79
720 RETURN
730 REM INVOER VRAGEN
740 PRINT CHR$(125);"HOEK:";
750 INPUT TH:IF TH<1 OR TH>89 THEN 740
760 PRINT CHR$(125);"SNELHEID (1-50)";
770 INPUT V:IF V<1 OR V>50 THEN 760
780 POKE 752,1:PRINT
790 RETURN

```

Slagbal

Dit spel is voor twee spelers en ieder moet een doel verdedigen tegen de tegenstander.

Elke speler heeft vijf ballen waarmee hij kan spelen en scoren, u kunt natuurlijk ook de bal van de tegenstander gebruiken.



Hoe te spelen

Elke speler heeft twee toetsen om zijn racket op en neer te laten bewegen.

SPELER 1: Toets A en Z voor op en neer.

SPELER 2: Toets + en / voor op en neer.

Om de moeilijkheidsgraad van het spel te veranderen, kunt u de maat van de rackets wijzigen door het getal in regel 50 aan te passen.

Helaas... het is niet mogelijk de maat van slechts één racket te wijzigen.

Programma

```

10 REM SLAGEBAL
20 POKE 752,1
30 P1=0:P2=0
40 G=10
50 H=4
60 Z=19-H
70 Q=5
80 SETCOLOR 2,7,10
90 REM SPEL ROUTINE
100 GOSUB 480
110 A=10:B=10
120 COLOR 2
130 PLOT 0,A:DRAWTO 0,A+H-1
140 PLOT 39,B:DRAWTO 39,B+H-1
150 C=0:D=0
160 IF G=0 THEN 730
170 IF G>5 THEN GOSUB 550:GOTO 190
180 GOSUB 580
190 REM SPELER ROUTINE
200 K=PEEK(764)
210 IF K=63 OR K=23 THEN C=K
220 IF K=38 OR K=6 THEN D=K
230 IF C=63 AND A>Q THEN GOSUB 610
240 IF C=23 AND A<Z THEN GOSUB 640
250 IF D=6 AND B>Q THEN GOSUB 670
260 IF D=38 AND B<Z THEN GOSUB 700
270 S=X+I:T=Y+J
280 IF S=-1 OR S=40 THEN 380
290 LOCATE S,T,U
300 IF U=2 THEN I=-I:GOTO 270
310 IF U=1 AND T=0 THEN J=-J:GOTO 270
320 IF U=1 AND T=23 THEN J=-J:GOTO 270
330 IF U=1 THEN I=-I:GOTO 270
340 COLOR 4:PLOT X,Y
350 COLOR 3:PLOT S,T
360 X=S:Y=T
370 GOTO 190
380 REM BAL WEG
390 IF S=-1 THEN P2=P2+1
400 IF S=40 THEN P1=P1+1
410 FOR L=0 TO 100 STEP 2
420 SOUND 0,L,10,14:SOUND 1,L+6,10,14
430 SOUND 2,L+12,10,14:SOUND 3,L+18,10,14
440 NEXT L
450 FOR L=0 TO 3:SOUND L,0,0,0:NEXT L
460 G=G-1
470 GOTO 90

```


```

480 REM VELD SPELEN
490 GRAPHICS 3+16
500 COLOR 1:PLOT 1,5:DRAWTO 1,0
510 DRAWTO 38,0:DRAWTO 38,5
520 PLOT 1,18:DRAWTO 1,23
530 DRAWTO 38,23:DRAWTO 38,18
540 RETURN
550 REM SERVICE SPELER EEN
560 X=2:Y=12:I=1:J=1
570 RETURN
580 REM SERVICE SPELER TWEЕ
590 X=37:Y=12:I=-1:J=1
600 RETURN
610 REM SPELER EEN OP
620 COLOR 2:A=A-1:PLOT 0,A
630 COLOR 4:PLOT 0,A+H:RETURN
640 REM SPELER EEN NEER
650 COLOR 2:PLOT 0,A+H:A=A+1
660 COLOR 4:PLOT 0,A-1:RETURN
670 REM SPELER TWEЕ OP
680 COLOR 2:B=B-1:PLOT 39,B
690 COLOR 4:PLOT 39,B+H:RETURN
700 REM SPELER TWEЕ NEER
710 COLOR 2:PLOT 39,B+H:B=B+1
720 COLOR 4:PLOT 39,B-1:RETURN
730 REM TIEN BALLEEN GEMIST
740 GRAPHICS 1+16:POSITION 1,8
750 PRINT #6;"- SPEL AFGELOPEN -"
760 POSITION 4,11
770 PRINT #6;"EEN      TWEЕ"
780 POSITION 5,17:PRINT #6;STR$(P1)
790 POSITION 13,17:PRINT #6;STR$(P2)
800 OPEN #1,4,0,"K:"
810 GET #1,A
820 IF A<>ASC(" ") THEN 810
830 CLOSE #1
840 RUN

```

Eivanger

Dit is uw kans om na te gaan of u ooit nog bruikbaar bent op een kippenboerderij. We gooien een eindeloze stroom eieren vanaf de bovenkant van het scherm, in een willekeurig patroon. U moet deze opvangen! Als u er eentje heeft gevangen, dient deze meteen naar een lift te worden gebracht die ze weer omhoog tilt. Dit levert u punten op. Trouwens, u kunt maar één ei in uw mandje hebben, u weet dat het niet goed is om meer eieren in een mand te hebben. Helaas!



Hoe te spelen

Gebruik de Z en X om de mand naar links en naar rechts te laten bewegen. Druk op de SPATIEBALK om onder het ei te stoppen.

U hoeft niet op andere toetsen te drukken als u de lift bereikt, uw ATARI zal de eieren automatisch overladen.

Laat u vijf eieren vallen dan is het spel uit. Tegen die tijd staat u tot aan uw knieën door een omelet te waden.

Programma


```

10 REM EIVANGER
20 DIM S$(5)
30 S$=" \_ / "
40 GRAPHICS 0:POKE 752,1:PRINT
50 SETCOLOR 2,4,0:SETCOLOR 1,0,10
60 FOR L=0 TO 18:POSITION 1,L:PRINT #6;"| |":NEXT L
70 B=5:POSITION B,20:PRINT #6;S$;
80 X=0:F=0:M=0:SC=0
90 POSITION B,20:PRINT #6;S$;
100 IF M=5 THEN 340
110 IF PEEK(764)=23 AND B>0 THEN B=B-1
120 IF PEEK(764)=22 AND B<36 THEN B=B+1
130 IF X<>0 THEN 160
140 IF RND(1)<0.98 THEN 220
150 X=5+INT(RND(1)*30):Y=0
160 LOCATE X,Y+1,H:IF H<>92 AND H<>47 AND H<>95 THEN
190
170 F=1:S$=" \0/ ":SC=SC+10
180 POSITION X,Y:PRINT #6;" ";X=0:GOTO 220
190 POSITION X,Y:PRINT #6;" ";
200 Y=Y+1:POSITION X,Y:PRINT #6;"0";
210 IF Y=22 THEN POSITION X,Y:PRINT #6;" ";X=0:M=M+1
220 IF B<>0 OR F<>1 THEN 90
230 S$=" \_ / "
240 F=0:POSITION 0,20:PRINT #6;S$;
250 FOR L=18 TO 0 STEP -1
260 SC=SC+1:SOUND 0,L*20,10,14
270 POSITION 2,L:PRINT #6;"0"
280 POSITION 2,L+1:PRINT #6;" ";
290 POSITION 15,23:PRINT #6;"SCORE : ";SC;
300 NEXT L
310 POSITION 2,0:PRINT #6;" ";
320 SOUND 0,0,0,0
330 GOTO 90
340 REM ALS ER VIJF ONTSNAPT ZIJN
350 GRAPHICS 1+16
360 POSITION 0,8:PRINT #6;" U HEEFT ER VIJF
GEMIST"
370 POSITION 0,15:PRINT #6;" SCORE : ";SC
380 POSITION 0,20:PRINT #6;" <RETURN>"
390 OPEN #1,4,0,"K:"
400 GET #1,D:IF D<>155 THEN 400
410 CLOSE #1
420 RUN

```

Zwarte gaten

Deze keer komen twee wezens met zelfmoordneigingen op u af. U heeft zeer weinig tijd om ze te identificeren en te reageren, omdat ze de mogelijkheid hebben om in een zwart gat te verdwijnen. Hierin kunnen ze nog steeds van koers veranderen en kunnen ze opnieuw verschijnen op onverwachte plaatsen. U zult al uw snelheid en concentratie nodig hebben om hun baan te schatten.



Hoe te spelen

Gebruik de Z en X voor links en rechts en de SPATIEBALK om te vuren. Bij dit spel gaat het om vooruitzien, of anticipatie, dus kijk uit en kijk vooruit.

Programma

```

10 REM ZWARTE GATEN
20 REM MET M/CODE BEWEGING
30 DIM M$(20)
40 GOSUB 680
50 SC=0:LV=0
60 POKE 623,4
70 GRAPHICS 5:POKE 752,1:PRINT
80 GOSUB 590
90 GOSUB 740
100 X=120:POKE H3,X:POKE C3,200
110 A=INT(RND(1)*110+55):B=INT(RND(1)*110+55)
120 P=50:Q=50
130 POKE C1,40:POKE C2,74:POKE H1,A:POKE H2,B
140 M=0:HH=0:GOSUB 870
150 Z=USR(12288):Z=USR(12311):Z=USR(12288):Z=USR(1231
1)
160 POKE 53278,1
170 IF HH=2 THEN 420
180 P=P+2:Q=Q+4
190 IF A=0 THEN 230
200 A=A+INT(RND(1)*5-2)
210 IF P>100 THEN A=A+3*SGN(X-A):POKE H1,A
220 IF P>240 THEN A=INT(RND(1)*110+55):P=P-256:POKE H
1,A
230 IF B=0 THEN 270
240 B=B+INT(RND(1)*5-2)
250 IF Q>100 THEN B=B+3*SGN(X-B):POKE H2,B
260 IF Q>240 THEN B=INT(RND(1)*110+55):Q=Q-256:POKE H
2,B
270 IF PEEK(53262)<>0 THEN 500
280 IF PEEK(764)=23 AND X>45 THEN X=X-2:POKE H3,X
290 IF PEEK(764)=22 AND X<200 THEN X=X+2:POKE H3,X
300 IF PEEK(764)=33 AND M=0 THEN POKE H4,X:GOSUB 810:
M=190
310 IF M=0 THEN 410
320 POKE 53278,1
330 M=M-5:IF M<35 THEN M=0:POKE 764,0
340 Z=USR(12342):POKE C4,INT(RND(1)*256)
350 IF PEEK(53263)=0 THEN 410
360 GOSUB 980
370 IF PEEK(53263)=1 THEN HH=HH+1:SC=SC+10:A=0:POKE H
1,A:M=0:POKE H4,M:POKE 764,0:GOTO 400
380 IF PEEK(53263)=2 THEN HH=HH+1:SC=SC+20:B=0:POKE H
2,B:POKE 764,0:GOTO 400
390 HH=2:SC=SC+40:A=0:B=0:POKE H1,A:POKE H2,B:M=0:POK
E H4,M:POKE 764,0
400 PRINT "          SCORE : ";SC

```

```

410 GOTO 150
420 GRAPHICS 1+16:POKE H1,0:POKE H2,0:POKE H3,0:POKE
H4,0
430 POSITION 0,10
440 M$="DUS..ER DOOR GEKOMEN":GOSUB 820
450 FOR L=1 TO 400:NEXT L
460 POSITION 0,15
470 M$="EN PROBEER DEZE DAN.":GOSUB 820
480 FOR L=1 TO 400:NEXT L
490 LV=LV+1:GOTO 70
500 REM SPELER DOOD
510 GRAPHICS 1+16
520 POKE H1,0:POKE H2,0:POKE H3,0:POKE H4,0
530 POSITION 2,6:PRINT #6;" MISLUKT"
540 POSITION 2,10:PRINT #6;" SCORE :";SC
550 POSITION 2,15:PRINT #6;" <return>"
560 OPEN #1,4,0,"K:"
570 GET #1,D:IF D<>155 THEN 570
580 CLOSE #1:GOTO 50
590 REM SPELERS OPZETTEN
600 POKE 559,62:POKE 53277,3
610 FMBASE=40:POKE 54279,FMBASE
620 H1=53248:H2=H1+1:H3=H2+1:H4=H3+1
630 C1=704:C2=C1+1:C3=C2+1:C4=C3+1
640 POKE H1,0:POKE H2,0:POKE H3,0:POKE H4,0
650 P1=FMBASE*256+1024:P2=P1+256:P3=P2+256:P4=P3+256
660 Z=USR(12357)
670 RETURN
680 REM M/C INLEZEN
690 RESTORE 1000
700 FOR L=12288 TO 12389
710 READ D:POKE L,D
720 NEXT L
730 RETURN
740 REM SPELERS POSITIES
750 RESTORE 1100
760 FOR L=184 TO 191:READ D:POKE P3+L,D:NEXT L
770 FOR L=50 TO 57:READ D
780 POKE P1+L,D:POKE P2+L,D
790 NEXT L
800 RETURN
810 Z=USR(12378):FOR L=176 TO 183:POKE P4+L,24:NEXT L
:RETURN
820 FOR L=1 TO LEN(M$)
830 PRINT #6;M$(L,L);
840 FOR W=1 TO 30:NEXT W
850 NEXT L
860 RETURN
870 REM SPEELVELD TEKEKEN
880 DP=LV*3

```

```


890 IF DP>25 THEN DP=25
900 COLOR 1:SETCOLOR 0,INT(RND(1)*15+1),INT(RND(1)*10
+5)
910 FOR L=6 TO DP+12
920 PLOT 0,L:DRAWTO 79,L
930 NEXT L
940 COLOR 4
950 FOR L=1 TO 20
960 PLOT INT(RND(1)*80),INT(RND(1)*(DP+6)+6)
970 NEXT L:RETURN
980 FOR EX=0 TO 200 STEP 4:SOUND 0,EX,4,14:NEXT EX:SO
UND 0,0,0,0:RETURN
990 REM
1000 DATA 173,255,44,72,162,254,189,0,44,157,1,44,202
,224,255,208,245
1010 DATA 104,141,0,44,104,96
1020 DATA 173,254,45,72,173,255,45,72,162,253,189,0,4
5,157,2,45,202,224,255
1030 DATA 208,245,104,141,1,45,104,141,0,45,104,96
1040 DATA 162,0,189,6,47,157,0,47
1050 DATA 232,224,248,208,245,104,96
1060 DATA 162,0,169,0,157,0,44,157,0,45,157,0,46,157,
0,47
1070 DATA 232,208,241,104,96
1080 DATA 162,0,169,0,157,0,47,232
1090 DATA 208,250,104,96
1100 DATA 0,0,24,24,24,60,126,255
1110 DATA 195,102,102,126,231,60,24,24
  
```

Meteorietenstorm

Uw verdedigingsscherm is buiten werking, uw kanonnen zijn oververhit en u nadert een meteorietenstorm.

We hebben u niet veel hoop gegeven, maar het is een kwestie van tijd en manuele vaardigheid, omdat het er om gaat hoe lang u kunt overleven.

Uw cursor beweegt in neerwaartse richting op het scherm en een ononderbroken stroom stenen en ster-overblijfselen reist u tegemoet.



Hoe te spelen

Omdat u geen verdedigingsscherm heeft, moet u proberen in leven te blijven door naar links en naar rechts uit te wijken. Dit doet u door gebruik van de Z- en X-toetsen. Om de zijwaartse beweging te stoppen, drukt u op de SPATIEBALK.

Om u te helpen langer in leven te blijven, hebben we u drie levens gegeven. U zult ze nodig hebben!

Programmeertips

Om de meteorieten langzamer te laten bewegen, kunt u een regel toevoegen, als volgt:

```
65 FOR Z = 1 TO 30 : NEXT Z
```

Programma

```

10 REM METEORIETEN STORM
20 GRAPHICS 0:POKE 752,1:PRINT
30 SETCOLOR 2,0,0
40 LI=3:SC=0
50 GOSUB 300
60 FOR Z=1 TO 30:NEXT Z
70 SC=SC+2
80 LOCATE X+1,0,A:LOCATE X+2,0,B:LOCATE X+3,0,C
90 IF A+B+C=96 THEN 140
100 LI=LI-1:GOSUB 330
110 GRAPHICS 0:POKE 752,1:PRINT
120 IF LI<>0 THEN SETCOLOR 2,0,0:GOTO 50
130 GOTO 180
140 POSITION X,0:PRINT " <=> ";
150 IF PEEK(764)=23 AND X>0 THEN X=X-1
160 IF PEEK(764)=22 AND X<35 THEN X=X+1
170 GOTO 50
180 REM VERNIETIGD
190 GRAPHICS 0:POKE 752,1:PRINT
200 POSITION 15,8:PRINT "U HEEFT"
210 GOSUB 460
220 POSITION 1,11:PRINT "DRIE ZEER KOSTBARE GALACTISC
HE SHUTTLES          TOTAAL VERNIETIGD"
230 GOSUB 460
240 POSITION 10,14:PRINT "DE REKENING VOLGT..."
250 GOSUB 460
260 POSITION 3,18:PRINT "UW SCORE IS: .....
....."
270 POSITION 15,18:PRINT SC
280 POKE 764,0
290 FOR T=1 TO 500:NEXT T:RUN
300 REM METEORIETEN PRINT EN BEWEEG
310 POSITION INT(RND(1)*40),23:PRINT "*"
320 RETURN
330 REM HOEVEEL LEVENS OVER
340 GOSUB 480
350 IF LI=0 THEN RETURN
360 GRAPHICS 0:POKE 752,1:PRINT
370 POSITION 1,5:PRINT "U HEEFT NOG ";LI;" GALACTISCH
E SHUTTLE(S)"
380 IF LI=0 THEN 450
390 POSITION 13,12
400 FOR L=1 TO LI
410 ? " <=> ";
420 NEXT L
430 SOUND 0,0,0,0
440 GOSUB 460


```

```
450 RETURN
460 FOR W=1 TO 500:NEXT W
470 RETURN
480 REM EXPLOSIE
490 FOR L=0 TO 255 STEP 3
500 SOUND 0,L,8,16-L/16
510 NEXT L
520 RETURN
```

Wegpiraat

U suist een berg af in een nieuwe sportauto (zeer indrukwekkend voor uw vriendin, die ook in de auto zit.) Uw remmen werken opeens niet meer en de weg wordt steeds steiler.

Houd de auto op de weg. U weet dat er heel wat van afhangt.



Hoe te spelen

Stuur de auto met behulp van de Z- en X-toetsen en druk op de SPATIEBALK als u de achterwielslip moet corrigeren. Als u eindelijk stilstaat, ziet u de score op het scherm.

Programma

```

10 REM WEGPIRAAT
20 GOSUB 300
30 GRAPHICS 0
40 POKE 752,1:PRINT
50 R=16
60 SETCOLOR 2,0,0
70 FOR L=0 TO 15:POSITION L,23:PRINT "+"
80 POSITION 39-L,23:PRINT "+":NEXT L
90 B=15
100 SC=0
110 GOSUB 400
120 SC=SC+1
130 POSITION B,0:PRINT "J=[ "
140 IF PEEK(764)=23 AND B>0 THEN B=B-1
150 IF PEEK(764)=22 AND B<37 THEN B=B+1
160 LOCATE B,1,E:LOCATE B+1,1,F:LOCATE B+2,1,G
170 IF E+F+G=96 THEN 110
180 REM IN DE BERM
190 FOR L=0 TO 255 STEP 3
200 SOUND 0,L,12,L/16
210 NEXT L
220 GRAPHICS 1+16
230 POSITION 5,8:PRINT #6;" uw auto"
240 POSITION 3,10:PRINT #6;"is total loss"
250 POSITION 1,15:PRINT #6;"UW SCORE WAS: ";SC
260 OPEN #1,4,0,"K:"
270 GET #1,D
280 IF D<>155 THEN 270
290 CLOSE #1:RUN
300 REM SCHERM INSTRUCTIES
310 GRAPHICS 1+16
320 POSITION 2,5:PRINT #6;"....WEGPIRAAT...."
330 POSITION 2,12:PRINT #6;"'spatie' RECHTDOOR"
340 POSITION 4,14:PRINT #6;"'z'     LINKS"
350 POSITION 4,16:PRINT #6;"'x'     RECHTS"
360 POSITION 0,20:PRINT #6;"return OM TE STARTEN"
370 OPEN #1,4,0,"K:"
380 GET #1,D:IF D<>155 THEN 380
390 CLOSE #1:RETURN
400 REM WEG TEKENEN & BEWEGEN
410 POSITION R,23:PRINT "+      +"
420 IF RND(1)>0.7 AND R>0 THEN R=R-1
430 IF RND(1)>0.7 AND R<33 THEN R=R+1
440 RETURN

```

Muurbreker



Dit is onze versie van het populaire spel 'BREAKOUT'. Aan de bovenkant van het scherm ziet u drie muren van verschillende kleuren en u moet de muren met behulp van een bal in zo weinig mogelijk slagen doorbreken.

Een kleine tip: Als de bal door de eerste twee muren heen is en achter de derde muur zit, zal hij daar heen en weer blijven kaatsen tot hij er weer uit kan.



Hoe te spelen

Gebruik de Z en X om de racket te laten bewegen.

Programmeertips

Als u het spel te moeilijk of te gemakkelijk vindt, kunt u de grootte van de racket veranderen door de waarde achter BL te verhogen of te verlagen.

Als u achterover wilt zitten en de bal het werk wil laten doen, verhoog dan de racketmaat totdat deze de hele onderkant in beslag neemt.

Programma

```

10 REM MUURBREKER (BREAKOUT)
20 GRAPHICS 7+16
30 MX=159:MY=95
40 GOSUB 590
50 X=10:Y=60
60 CX=2:CY=-1
70 SC=0:W=0
80 BX=10:BY=90
90 BL=19
100 COLOR 3:PLOT BX,BY:DRAWTO BX+BL,BY
110 TX=X+CX:TY=Y+CY
120 IF SC=1950 THEN W=W+1:SC=0:GOSUB 590
130 LOCATE TX,TY,D
140 IF TX<2 OR TX>MX-2 THEN GOSUB 250:GOTO 110
150 IF TY<1 THEN GOSUB 280:GOTO 110
160 IF D<>0 THEN GOSUB 310:GOTO 110
170 IF TY>93 THEN GOTO 450
180 COLOR 4:PLOT X,Y
190 COLOR 3:PLOT TX,TY
200 X=TX:Y=TY
210 K=PEEK(764)
220 IF K=22 AND BX+BL<MX-3 THEN GOSUB 390
230 IF K=23 AND BX>3 THEN GOSUB 420
240 GOTO 110
250 REM MUUR GERAAKT
260 SOUND 1,81,10,12:CX=-CX
270 SOUND 1,0,0,0:RETURN
280 REM BOVENMUUR OF RACKET GERAAKT
290 SOUND 1,126,10,12:CY=-CY
300 SOUND 1,0,0,0:RETURN
310 REM RACKET OF STEEN GERAAKT
320 IF D=3 THEN 360
330 SOUND 1,63,10,15:COLOR 4:SOUND 1,0,0,0:Z=2*INT(TX
/2)
340 PLOT Z,TY:PLOT Z+1,TY:SC=SC+D*10
350 CY=-CY:RETURN
360 REM RACKET GERAAKT
370 SOUND 1,40,10,14:CY=-CY:SOUND 1,0,0,0
380 SETCOLOR 2,INT(RND(1)*16),15:RETURN
-390 REM RACKET LINKS
400 COLOR 4:PLOT BX,BY:PLOT BX+1,BY
410 BX=BX+2:COLOR 3:PLOT BX+BL-1,BY:PLOT BX+BL,BY:RET
URN
-420 REM RACKET RECHTS
430 COLOR 4:PLOT BX+BL,BY:PLOT BX+BL-1,BY
440 BX=BX-2:COLOR 3:PLOT BX,BY:PLOT BX+1,BY:RETURN
450 REM BAL IS WEG

```

```

460 FOR L=0 TO 240 STEP 3
470 SOUND 0,L,10,10;SOUND 1,L,10,10
480 SOUND 2,L+8,10,6;SOUND 3,L+15,10,8
490 NEXT L
500 FOR L=0 TO 3;SOUND L,0,0,0;NEXT L
510 GRAPHICS 1+16;POSITION 0,10;PRINT #6;"*****
*****"
520 PRINT #6;"          SCORE : ";STR$(SC+W*2000);PRINT #
6;PRINT #6;"*****"
530 POSITION 0,19;PRINT #6;"          <RETURN>"
540 OPEN #1,4,0,"K:"
550 GET #1,A
560 IF A<>155 THEN 550
570 SOUND 1,63,10,15;SOUND 1,0,0,0
580 CLOSE #1;GOTO 10
590 REM MUREN EN STENEN TEKENEN
600 SETCOLOR 2,2,10;SETCOLOR 0,7,4
610 SETCOLOR 1,6,8;COLOR 3
620 PLOT 0,0;DRAWTO MX,0
630 PLOT 0,0;DRAWTO 0,MY
640 PLOT 1,0;DRAWTO 1,MY
650 PLOT MX,0;DRAWTO MX,MY
660 PLOT MX-1,0;DRAWTO MX-1,MY
670 FOR L=2 TO MX-5 STEP 4
680 COLOR 1;PLOT L,20;PLOT L+1,20
690 COLOR 2;PLOT L,12;PLOT L+1,12
700 PLOT L,28;PLOT L+1,28
710 NEXT L
720 RETURN

```


Oceaanslag



Hier volgt een oud commando van de jongens van het Ministerie van Defensie.
Uw vijand ligt aan de andere kant van de horizon en is zelfs niet gewaarschuwd dat u op het punt staat om aan te vallen, dus het verrassingselement ligt geheel aan uw kant.

Uw vijand heeft een grote vloot en is onderweg om u aan te vallen, dus u bent hem eigenlijk voor.

Trouwens, het is maar een spel hoor.



Hoe te spelen

De vijandelijke vloot bestaat uit:

- 1 vlaggeschip – 240 punten
- 1 kruiser – 160 punten
- 1 fregat – 100 punten

U heeft slechts een paar torpedo's, dus laat ze wel werken!

Gebruik de Z en X om de torpedobuizen naar links en rechts te laten draaien en de / en + om ze op en neer te laten bewegen.

Als u denkt de vijand te zien, druk dan op de SPATIEBALK om te schieten.

Programmeertips

Om de mogelijkheden van het spel iets in te krimpen en daardoor hogere eisen te stellen aan de speler, kunt u het aantal torpedo's verkleinen door de waarde van G in regel 60 te veranderen.

Programma

```

10 REM OCEAANSLAG
20 DIM A(9,9)
30 FOR X=0 TO 9:FOR Y=0 TO 9
40 A(X,Y)=0:NEXT Y:NEXT X
50 GOSUB 1000
60 SC=0:G=40
70 GOSUB 1500:GOSUB 2000
80 X=5:Y=5:COLOR 2
90 PLOT X,10:PLOT 10,Y
100 REM HOOFDLUS
110 IF SC=500 THEN 500
120 IF G=0 THEN 400
130 GET #1,D
140 IF D=32 THEN 300
150 IF D=90 AND X>0 THEN COLOR 4:PLOT X,10:COLOR 2:X=
X-1:PLOT X,10
160 IF D=88 AND X<9 THEN COLOR 4:PLOT X,10:COLOR 2:X=
X+1:PLOT X,10
170 IF D=47 AND Y<9 THEN COLOR 4:PLOT 10,Y:COLOR 2:Y=
Y+1:PLOT 10,Y
180 IF D=43 AND Y>0 THEN COLOR 4:PLOT 10,Y:COLOR 2:Y=
Y-1:PLOT 10,Y
190 REM
200 GOTO 100
300 REM SCHIETEN
310 G=G-1
320 IF A(X,Y)=0 THEN PRINT "GEMIST...";G;" SCHOTEN":C
OLOR 4:PLOT X,Y:A(X,Y)=-1:GOTO 100
330 IF A(X,Y)=-1 THEN PRINT "O.K.  WEER GEMIST!":GOT
O 100
340 ON A(X,Y) GOSUB 4000,4100,4200
350 COLOR 1:PLOT X,Y:PRINT "***** BOEM ***** ";G;" SC
HOTEN":A(X,Y)=4:GOTO 100
400 REM GEEN SCHOTEN MEER
410 PRINT "ZE WAREN ZO VERSTOPT..."
420 GOSUB 3000
430 GOTO 540
500 REM ALLE SCHEPEN GEZONKEN
510 PRINT "GEFELICITEERD !"

```



```
520 PRINT "ALLES GEZONKEN MET ";G;" SCHOTEN OVER"
530 FOR W=1 TO 500:NEXT W
540 PRINT "NOGMAALS (J/N) ":POKE 764,255
550 IF PEEK(764)=1 THEN POKE 764,255:RUN
560 GOTO 550
570 END

1000 REM SCORE PRINTEN
1010 GRAPHICS 1+16:POSITION 0,3
1020 PRINT #6;"SLAGSCHIP SCORES"
1030 POSITION 0,7
1040 PRINT #6;" 0 FREGAT          100"
1050 POSITION 0,11
1060 PRINT #6;"000 KRUISER       160"
1070 POSITION 0,12
1080 PRINT #6;" 0"
1090 POSITION 0,14
1100 PRINT #6;"000"
1110 POSITION 0,15
1120 PRINT #6;"0 0 VLAGGESCHIP  240"
1130 POSITION 0,16
1140 PRINT #6;"000"
1150 OPEN #1,4,0,"K:"
1160 GET #1,D:IF D<>32 THEN 1160
1170 RETURN

1500 REM SPEELVELD TEKENEN
1510 GRAPHICS 3
1520 POKE 752,1:COLOR 3
1530 FOR X=0 TO 9
1540 PLOT X,0:DRAWTO X,9
1550 NEXT X
1560 RETURN

2000 REM POSITIES KIEZEN
2010 A=INT(RND(1)*8):B=INT(RND(1)*8)
2020 FOR L=0 TO 2:A(A+L,B)=3
2030 A(A+L,B+2)=3:A(A,B+L)=3
2040 A(A+2,B+L)=3:NEXT L
2050 A=INT(RND(1)*10):B=INT(RND(1)*10)
2060 IF A(A,B)<>0 THEN 2040
2070 A(A,B)=1
2080 A=INT(RND(1)*8):B=INT(RND(1)*9)
2090 IF A(A,B)<>0 OR A(A+1,B)<>0 OR A(A+2,B)<>0 OR A(A+1,B+1)<>0 THEN 2080
2100 A(A,B)=2:A(A+1,B)=2:A(A+2,B)=2
2110 A(A+1,B+1)=2
2120 RETURN

3000 REM LAAT ZIEN WAAR ZE ZIJN
3010 COLOR 1
3020 FOR X=0 TO 9
3030 FOR Y=0 TO 9
3040 IF A(X,Y)>0 THEN PLOT X,Y
```




```
3050 NEXT Y
3060 NEXT X
3070 RETURN
4000 SC=SC+100:RETURN
4100 SC=SC+40:RETURN
4200 SC=SC+30:RETURN
```

Wegversperring

U bent bij de Berlijnse muur met een bus vol passagiers om naar de andere kant te gaan, maar de grenswachten maken nieuwe wegversperringen nog sneller dan u kunt manoeuvreren.

Maakt u een fout en raakt u één van de versperringen dan is het afgelopen voor iedereen.



Hoe te spelen

U komt links boven op het scherm en u wordt verrast door een snel groeiend aantal versperringen. U beweegt met de toetsen Z en X voor links en rechts en de +/- en +/- toetsen voor op en neer. Als u besluit van het scherm af te rijden, merkt u dat dit niet kan, het scherm is namelijk aan drie kanten afgeschermd.

Programmeertips

Als u het gemakkelijker wilt maken, kunt u de lengte van de bus veranderen door de waarde van N in regel 30 te modificeren naar een cijfer tussen 1 en 20.

Programma

```

10 REM WEGVERSERRING
20 DIM X(20),Y(20)
30 N=4
40 GRAPHICS 3+16
50 FOR L=1 TO N:X(L)=0:Y(L)=0:NEXT L
60 SC=0
70 REM HOOFDRoutine
80 A=X(1):B=Y(1)
90 K=PEEK(764):IF K=23 OR K=22 OR K=6 OR K=38 THEN M=
K:SC=SC+10
100 IF M=23 THEN A=A-1:IF A=-1 THEN A=39
110 IF M=22 THEN A=A+1:IF A=40 THEN A=0
120 IF M=38 THEN B=B+1:IF B=24 THEN B=0
130 IF M=6 THEN B=B-1:IF B=-1 THEN B=23
140 LOCATE A,B,D:IF D<>0 THEN 270
150 COLOR 1:PLOT A,B
160 COLOR 2:PLOT X(1),Y(1)
170 COLOR 1>4:PLOT X(N),Y(N)
180 FOR L=N-1 TO 1 STEP -1
190 X(L+1)=X(L):Y(L+1)=Y(L)
200 NEXT L
210 X(1)=A:Y(1)=B
220 G=INT(RND(1)*40):H=INT(RND(1)*24)
230 LOCATE G,H,I
240 IF I=1 OR I=2 THEN 220
250 COLOR 3:PLOT G,H
260 GOTO 70
270 REM EEN VERSERRING GERAAKT
280 GRAPHICS 1+16
290 POSITION 3,10:PRINT #6;"U SCORDE :";SC
300 POSITION 5,17:PRINT #6;"<SPATIE>"
310 OPEN #1,4,0,"K:"
320 GET #1,D
330 IF D<>32 THEN 320
340 CLOSE #1
350 RUN

```


Mijnenveld



Het is een moeilijke opdracht voor ons soldaatje, om zijn weg te vinden uit het mijnenveld. Zullen we zien of we hem kunnen helpen de weg terug te vinden naar zijn collega's.

De mijnen zijn willekeurig gelegd, dus is het oppassen met elke stap die uw soldaatje waagt. We zullen het onze held maar niet vertellen, maar wij hebben het uit betrouwbare bron dat het wellicht het veiligste is een andere opdracht aan te vragen.

Telkens als hij door een mijnenveld komt, staat hij in een ander met een nog groter aantal mijnen. Hij moet doorgaan tot ...



Hoe te spelen

Gebruik de Z en X voor links en rechts en / en + voor op en neer. Wanneer het onvermijdelijke gebeurt, zal de score op het scherm verschijnen. U kunt op de RETURN-toets drukken om uw kleine soldaat terug te sturen naar het mijnenveld.

Programmeertips

Als u denkt dat onze held een pauze verdient, kunt u het aantal mijnen verlagen door de waarde van DI = 10 in regel 30 te veranderen.

Dit zal de tussenpozen waarmee de mijnen worden gelegd en de reactietijd verlengen.

Programma

```

10 REM MIJNEN VELD MET P/M IN M/C
20 GRAPHICS 5+16
30 DI=10:REM MOEILIJKHEIDSGRAAD
40 GOSUB 1500:GOSUB 1000
50 M=70:POKE H1,M
60 N=120:POKE C1,200
70 SC=0:WV=1
80 X=20:Y=20:GOSUB 5000
90 POKE 53278,1
100 IF PEEK(764)=23 AND M>45 THEN M=M-2:POKE H1,M
110 IF PEEK(764)=22 AND M<205 THEN M=M+2:POKE H1,M
120 IF PEEK(764)=6 AND N>25 THEN N=N-3:Z=USR(12300)
130 IF PEEK(764)=38 AND N<220 THEN N=N+3:Z=USR(12339)
140 IF PEEK(53252)=0 THEN 200
150 IF PEEK(53252)=2 THEN 400
160 GOTO 800
200 GOTO 100
400 REM EINDE OPDRACHT/DOEL GEVONDEN
410 FOR L=0 TO 250 STEP 3:SOUND 0,L,10,14
420 NEXT L
430 SOUND 0,0,0,0
440 GOSUB 5000:SC=SC+WV*10:WV=WV+1
450 GOTO 100
800 REM MIJN GERAAKT
810 REM EN KNALLEN..
820 FOR L=0 TO 255 STEP 2
830 SOUND 0,L,8,L/32
835 SETCOLOR 2,L/16,L/16
840 NEXT L
850 SOUND 0,0,0,0
860 GRAPHICS 1+16:POKE H1,0
870 POSITION 0,6:PRINT #6;"U RAAKTE EEN MIJN"
880 POSITION 2,10:PRINT #6;"U SCOORDE : ";SC
890 POSITION 4,20:PRINT #6;"<RETURN>"
900 OPEN #1,4,0,"K:"
910 GET #1,D:IF D<>155 THEN 910
920 CLOSE #1
930 RUN
1000 REM SPELERS OPZETTEN
1010 POKE 559,62:POKE 53277,3
1020 PMBASE=40:POKE 54279,PMBASE
1030 H1=53248
1040 C1=704
1050 POKE H1,0
1060 P1=PMBASE*256+1024
1070 Z=USR(12288)
1080 RESTORE 7000

```

```

1090 FOR L=0 TO 7
1100 READ D:POKE P1+L+120,D
1110 NEXT L
1120 RETURN
1500 REM MACHINETAAL INLEZEN
1510 RESTORE 9000
1520 FOR L=12288 TO 12377
1530 READ D:POKE L,D
1540 NEXT L
1550 RETURN
5000 REM DOEL WEGHALEN/TEKENEN
5010 COLOR 4:GOSUB 5500
5020 X=INT(RND(1)*70+5):Y=INT(RND(1)*40+5)
5030 COLOR 2:GOSUB 5500
5100 FOR L=1 TO DI
5110 POKE 53278,1
5120 P=INT(RND(1)*70+5):Q=INT(RND(1)*40+5)
5130 COLOR 3:GOSUB 5600
5140 IF PEEK(53252)<>0 THEN COLOR 4:GOSUB 5600:GOTO 5
110
5150 NEXT L
5160 RETURN
5500 PLOT X-1,Y-1:DRAWTO X+1,Y-1
5510 DRAWTO X+1,Y+1:DRAWTO X-1,Y+1
5520 DRAWTO X-1,Y-1
5530 RETURN
5600 PLOT P-1,Q:DRAWTO P+1,Q
5610 PLOT P,Q-1:DRAWTO P,Q+1
5620 RETURN
7000 DATA 56,186,146,124,16,40,68,68
9000 DATA 162,0,169,0,157,0,44
9010 DATA 232,208,250,104,96
9100 DATA 173,2,44,72,173,1,44,72,173,0,44,72,162,0
9110 DATA 189,3,44,157,0,44,232,224,253,208,245
9120 DATA 104,141,253,44,104,141,254,44,104,141,255,4
4,104,96
9200 DATA 173,255,44,72,173,254,44,72,173,253,44,72,1
62,252
9210 DATA 189,0,44,157,3,44,202,224,255,208,245
9220 DATA 104,141,0,44,104,141,1,44,104,141,2,44,104,
96


```

Schiettent



Op elke kermis is wel een schiettent. Om hiervoor alvast te oefenen, is hier een programma dat ongeveer dezelfde mogelijkheden biedt.

Wij laten echter geen eenden of kippen op het scherm verschijnen om op te schieten maar de cijfers 0 tot en met 9.



Hoe te spelen

De cijfers zullen van links naar rechts over het scherm rollen en als u ze niet raakt, komen ze aan de andere kant van het scherm weer op. Elk cijfer dat u raakt, zal uit het rijtje verdwenen zijn.

Gebruik de Z en X voor links en rechts en de SPATIEBALK om te vuren.

Elke speler heeft 10 schoten en een score zal op het scherm verschijnen als de tien schoten zijn verbruikt.

Programmeertips

Als de rij cijfers te snel voorbijgaat, kunt u ze langzamer instellen. Aldus:

```
425 FOR Z = 1 TO 40 : NEXT Z
```

Dit vertraagt de loopsnelheid van de rij cijfers.

Programma

```

10 REM SCHIETTENT
20 DIM A$(41),B$(41),T$(41)
30 GRAPHICS 0:POKE 752,1:PRINT
40 A$(1,40)="1  2  3  4  5  6  7  8  9  0
"
50 Q=0
60 Y=0
70 SC=0:B=10:POSITION B,21:PRINT "^";
80 SH=10
90 GOSUB 360
100 IF PEEK(764)=23 AND B>0 THEN B=B-1:POSITION B,22:
PRINT " ^ ";
110 IF PEEK(764)=22 AND B<36 THEN POSITION B,22:PRINT
" ^ ";:B=B+1
120 GOSUB 360
130 IF PEEK(764)=33 AND Y=0 THEN Y=21:X=B+1:SH=SH-1
140 IF Y=0 THEN 90
150 POSITION X,Y:PRINT " ";
160 Y=Y-1:IF Y<>0 THEN 190
170 IF SH=0 THEN 280
180 GOTO 90
190 LOCATE X,Y,D
200 IF D<>32 THEN 230
210 POSITION X,Y:PRINT "|";
220 GOTO 90
230 T=X+1
240 A$(T,T)=" ":SC=SC+D*5-240
250 POSITION 12,23:PRINT " SCORE : ";SC;" ";
260 Y=0:POKE 764,0
270 GOTO 90
280 REM KOGELS OP
290 GRAPHICS 1+16:POSITION 0,4
300 PRINT #6;"UW KOGELS ZIJN OP"
310 POSITION 4,10:PRINT #6;"score :";SC
320 POSITION 4,17:PRINT #6;"< RETURN >"
330 OPEN #1,4,0,"K:"
340 GET #1,D:IF D<>155 THEN 340
350 RUN
360 REM STRINGS PRINTEN
370 T%=A$(40,40)
380 B$(2,40)=A$(1,39)
390 A$(1,1)=T%:A$(2,40)=B$(2,40)
400 POSITION 0,10:PRINT A$
410 Q=Q+1
420 IF Q=40 THEN Q=0:SETCOLOR 2,INT(RND(1)*16),0
430 RETURN


```

Anagrammen



Als u moe bent van het schieten en het vernietigen van objecten vanuit de ruimte, wilt u misschien op een andere wijze uw denkvermogen benutten.

Uw computer zal uit een woordenschat willekeurige woorden kiezen en de letters hiervan door elkaar op het scherm tonen. Denkt u het antwoord te weten, typ het dan in.



Hoe te spelen

Typ het woord dat door de letters wordt gevormd in en druk op RETURN. Er worden tien woorden gegeven en u krijgt punten gebaseerd op uw antwoorden.

Programmeertips

De woordenschat van de computer kan worden veranderd door de DATA in de regels 510 en hoger te veranderen. Vergeet niet dat als u het aantal woorden verandert, het DATA-getal in regel 500 ook veranderd zal moeten worden.

Programma

```

10 REM ANAGRAMMEN
20 DIM S$(20),W$(20),I$(20),T$(1),H$(20)
30 GRAPHICS 0
40 SETCOLOR 2,0,0
50 SC=0
60 FOR T=1 TO 10
70 H$=""
80 GOSUB 350:GOSUB 400
90 GRAPHICS 0:SETCOLOR 2,0,0
100 POSITION 11,4:PRINT "ANAGRAM NUMMER ";T
110 POSITION 5,12:PRINT "WAT IS ...."
120 POSITION 16,14:PRINT S$
130 POKE 755,3
140 POSITION 0,20:PRINT "
"
150 POSITION 16,20:PRINT H$
160 POSITION 15,12
170 INPUT I$
180 IF LEN(I$)=0 THEN 140
190 POKE 752,1:PRINT
200 IF I$=W$ THEN 240
210 POKE 755,2:POSITION 13,22
220 PRINT " WAS HET WOORD";
230 GOTO 260
240 POSITION 0,16:PRINT " GOED !!!!!.. U SCOORT 10 PU
NTEN"
250 SC=SC+10:POKE 755,2
260 FOR W=1 TO 700:NEXT W
270 NEXT T
280 PRINT CHR$(125)
290 POSITION 15,10:PRINT "RESULTATEN"
300 POSITION 13,14:PRINT "SCORE.....: ";SC
310 POSITION 15,20:PRINT "< SPATIE >"
320 POKE 764,0
330 IF PEEK(764) <> 33 THEN 330
340 RESTORE :PRINT CHR$(125):GOTO 40
350 REM EEN WOORD WILLEKEURIG KIEZEN
360 RESTORE 500
370 READ N
380 FOR L=1 TO INT(RND(1)*N+1)
390 READ W$:NEXT L:RETURN
400 REM WOORD IN W$ STOPPEN
410 S$=W$:FOR L=1 TO 30
420 A=INT(RND(1)*LEN(S$)+1)
430 B=INT(RND(1)*LEN(S$)+1)
440 T$=S$(A,A):S$(A,A)=S$(B,B)
450 S$(B,B)=T$:NEXT L

```

```
460 FOR Q=1 TO LEN(W$)
470 QQ=ASC(W$(Q,Q))+128
480 H$(Q,Q)=CHR$(QQ)
490 NEXT Q:RETURN
500 DATA 20
510 DATA ARTISTIEK,INTELEKTUEEL
520 DATA VRIJSTELLING,NIETIG
530 DATA TELEVISIE,MECHANISCH
540 DATA LUCHT,GEBORSTELD
550 DATA FORMATIE,KLOK
560 DATA EDUKATIE,VIDEO
570 DATA COMPUTER,AUTOMATISCH
580 DATA SYMBOLISCH,GEREGISTREERD
590 DATA MAAIVELD,TERRARIUM
600 DATA KROKODIL,NAALDHOUT
```

Eeuwkalender



Dit is uw kans om de computer u alles te laten vertellen: niet over de dagen van de week, maar over de dagen van de eeuw!

Als u uw familie, kennissen of vrienden vraagt naar hun verjaardag, en deze datum intoetst, zal dit programma u exact kunnen vertellen op welke dag van de week deze grote gebeurtenis plaatsvond.

Trouwens, de gegevens in dit programma zijn alleen geldig vanaf 1 januari 1753, toen de dag- en maand-telling z'n huidige vorm kreeg.

O, wist u dat niet?

We hebben trouwens wél rekening gehouden met schrikkeljaren ... voordat u denkt ons te slim af te zijn.



Hoe te gebruiken

U zal worden gevraagd om de datum die u in gedachten heeft in te toetsen. Drukt u wel op de RETURN-toets na ieder dag-, maand- en jaar-getal. Als alles is ingevoerd, zal uw ATARI XL computer de berekening voor u uitvoeren en u het resultaat laten zien.

Programma

```

10 REM EEUW KALENDER
20 DIM M$(20),T$(20)
30 GRAPHICS 0:POKE 710,144:POKE 712,144:POKE 82,8:PRINT
40 PRINT :PRINT "INVOER VAN DATUM:"
50 GOSUB 160
60 GOSUB 320
70 POSITION 4,2
80 PRINT "DAT IS/WAS/WORDT EEN ";T$
90 POKE 752,1:PRINT
100 POSITION 4,3:PRINT "DRUK OP SPATIE OM DOOR TE GAAN"
110 OPEN #1,4,0,"K:"
120 GET #1,D:IF D<>32 THEN 120
130 RESTORE
140 CLOSE #1
150 RUN
160 REM DATUM INVOER
170 POSITION 8,10:PRINT "JAAR .....":
POSITION 19,10
180 INPUT Y:IF Y<1753 OR Y<>INT(Y) THEN 170
190 POSITION 8,15:PRINT "MAAND.....":
POSITION 19,15
200 INPUT M$:IF M$(1,1)>="1" AND M$(1,1)<="9" THEN M=
VAL(M$):GOTO 270
210 IF LEN(M$)<3 THEN 190
220 RESTORE 500
230 FOR L=1 TO 12
240 READ T$
250 IF T$(1,3)=M$(1,3) THEN M=L
260 NEXT L
270 IF M<1 OR M>12 OR M<>INT(M) THEN 200
280 POSITION 8,20:PRINT "DAG (1-31)":
POSITION 19,20
290 INPUT D
300 IF D<1 OR D>31 OR D<>INT(D) THEN 280
310 RETURN
320 REM DAG VAN WEEK GEVONDEN
330 A=Y+1+31*(M-1)+INT((Y-1)/4)-INT(3*INT((Y+99)/100)
/4)
340 IF M<=2 THEN 360
350 A=Y+1+31*(M-1)-INT((4*M+23)/10)+INT(Y/4)-INT(3*(I
NT(Y/100)+1)/4)
360 A=A-INT(A/7)*7
370 IF A=0 THEN A=7
380 A=A-1:IF M=2 THEN 420
390 N=31

```




```
400 IF M=4 OR M=6 OR M=9 OR M=11 THEN N=30
410 GOTO 440
420 N=28
430 IF Y-INT(Y/4)*4=0 THEN N=29
440 IF D>N THEN T$="GEEN-DAG":RETURN
450 A=A+D:D=A-INT(A/7)*7:IF D=0 THEN D=7
460 RESTORE 540
470 FOR L=1 TO D
480 READ T$
490 NEXT L:RETURN
500 DATA JAN,FEB,MAA
510 DATA APR,MEI,JUN
520 DATA JUL,AUG,SEP
530 DATA OKT,NOV,DEC
540 DATA ZONDAG,MAANDAG,DINSDAG
550 DATA WOENSDAG,DONDERDAG,VRIJDAG
560 DATA ZATERDAG
```

Surround

Wel eens geprobeerd iemand in uw eentje te omsingelen? Als de tegenpartij tegelijkertijd hetzelfde probeert, kan het een hachelijke onderneming zijn.

Dit is uw kans om dit uit te proberen.

Probeer te vermijden dat u in uw eigen valletjes trapt en dat u uw eigen spoor kruist.



Hoe te spelen

Voor dit juweeltje van programmeertechniek hebben we alle instructies bij het begin van het spel op het scherm gezet.

Programma

```


5 REM SURROUND
10 DIM M$(12),N1$(12),N2$(12)
15 GOSUB 175:GOSUB 270
20 GRAPHICS 5+16
25 E=22:F=34
30 A=20:C=60
35 B=24:D=24
40 REM SPELER 1
45 K=PEEK(764):IF K=47 OR K=23 OR K=22 OR K=63 THEN E
=K
50 IF E=23 THEN A=A+79
55 IF E=22 THEN A=A+81
60 IF E=47 THEN B=B+47
65 IF E=63 THEN B=B+49
70 A=A-80*INT(A/80):B=B-48*INT(B/48)
75 K=PEEK(764):IF K=14 OR K=34 OR K=38 OR K=6 THEN F=
K
80 IF F=34 THEN C=C+79
85 IF F=38 THEN C=C+81
90 IF F=14 THEN D=D+47
95 IF F=6 THEN D=D+49
100 C=C-80*INT(C/80):D=D-48*INT(D/48)
105 LOCATE A,B,G:IF G<>0 THEN 130
110 LOCATE C,D,H:IF H<>0 THEN 140
115 COLOR 1:PLOT A,B
120 COLOR 2:PLOT C,D
125 GOTO 40
130 REM SPELER 1 STUK
135 M$=N1$:GOTO 150
140 REM SPELER 2 STUK
145 M$=N2$
150 GRAPHICS 1+16
155 POSITION 0,10:PRINT #6;M$;" IS STUK"
160 OPEN #1,4,0,"K:"
165 GET #1,D:IF D<>32 THEN 165
170 CLOSE #1:RUN
175 REM INSTRUKTIES
180 GRAPHICS 1+16
185 PRINT #6;" INSTRUKTIES (J/N)"
190 OPEN #1,4,0,"K:"
195 GET #1,D:IF D<>ASC("J") AND D<>ASC("N") THEN 195
200 IF D=ASC("N") THEN 265
205 POSITION 0,3:PRINT #6;"....INSTRUKTIES...."
210 POSITION 0,5:PRINT #6;"SPELER      SPELER"
215 POSITION 1,6:PRINT #6;"EEN      TWE"
220 POSITION 3,8:PRINT #6;"Z      LINKS  ."
225 POSITION 3,10:PRINT #6;"X      RECHTS / "
```

```
230 POSITION 3,12:PRINT #6;"Q      OP      -  "
235 POSITION 3,14:PRINT #6;"A      NEER    +  "
240 POSITION 0,16:PRINT #6;"probeer in leven te":PRIN
T #6:PRINT #6;"      blijven"
245 POSITION 2,20:PRINT #6;"SLUIT ELKAAR IN"
250 POSITION 0,22:PRINT #6;"spatie OM TE STARTEN"
255 GET #1,D:IF D<>32 THEN 255
260 CLOSE #1
265 CLOSE #1:RETURN
270 PRINT CHR$(125):POKE 710,0:POKE 752,1
275 POSITION 0,10:PRINT "NAAM SPELER EEN ":POSITION 1
8,10:? "
280 POSITION 18,10:INPUT N1$:IF N1$="" THEN 275
285 POSITION 0,10:PRINT "NAAM SPELER TWEE":POSITION 1
8,10:? "
290 POSITION 18,10:INPUT N2$:IF N2$="" THEN 285
295 IF N1$<>N2$ THEN 315
300 POSITION 0,5:PRINT "FASCINEREND..."
305 FOR L=1 TO 500:NEXT L
310 GOTO 275
315 RETURN
```

Hark the Herald Angels Sing

In een goede bui vonden we het leuk om in plaats van een laatste David Bowie hit een kerstlied te maken.

Om het nog spannender te maken, hebben we dit programma zó ontworpen dat de woorden van het lied op het scherm komen. Noem het 'Zing met uw computer mee' als u dat wilt.



Hoe te spelen

Als u het programma heeft ingetoetst, kunt u in een lekkere stoel gaan zitten en genieten van de goede muziek.

Programmeertips

Mocht u onze keuze niet leuk vinden, dan is het mogelijk de muziek te veranderen door de DATA in regel 470 en verder te wijzigen. De woorden staan als DATA in de regels 350 t/m 450. Onthoud dat het laatste 'woord' een punt (.) dient te zijn. Het DATA-getal in regel 460 vertelt de computer het totale aantal noten in het lied.

Programma

```

10 REM HARK THE HERALD ANGELS SING
20 DIM T(6):RESTORE 460
30 READ T
40 FOR L=1 TO 6:T(L)=L*T:NEXT L
50 DIM W$(400),T$(40)
60 W$=" ":W$(400)=W$:W$(2)=W$(1)
70 GOSUB 260
80 GRAPHICS 0:POSITION 0,8
90 POKE 752,1:PRINT
100 RESTORE 470
110 GOSUB 130
120 END
130 REM SPEEL EEN TOON
140 C=0
150 READ P,D
160 IF P<>-1 THEN 200
170 PRINT :PRINT W$(C*40+1,C*40+40)
180 C=C+1
190 GOTO 150
200 IF P=0 THEN 250
210 SOUND 0,P,10,14
220 FOR W=1 TO T(D):NEXT W
230 SOUND 0,0,0,0
240 GOTO 150
250 RETURN
260 REM WOORDEN INLEZEN
270 C=0
280 RESTORE 350
290 READ T$
300 IF T$(1,1)=",." THEN 340
310 W$(C*40+1,C*40+LEN(T$))=T$
320 C=C+1
330 GOTO 290
340 RETURN
350 DATA HOOR - DE ENG'LEN ZINGEN D'EER
360 DATA VAN DE NIEUW GEBOREN HEER
370 DATA VREED OP AARDE..'TIS VERVULD
380 DATA GOD VERZOENT DER MENSEN SCHULD
390 DATA VOEGT U VOLKEN IN HET KOOR
400 DATA DAT WEERKLINKT DE HEMEL DOOR
410 DATA ZINGT MET ALGEMENE STEM
420 DATA VOOR HET KIND VAN BETHLEHEM
430 DATA HOOR - DE ENG'LEN ZINGEN D'EER
440 DATA VAN DE NIEUW GEBOREN HEER
450 DATA .
460 DATA 80
  
```



470 DATA -1,0,108,2,81,2,81,3,85,1,81,2,64,2,64,2,72,
2
480 DATA -1,0,53,2,53,2,53,3,61,1,64,2,72,2,64,4
490 DATA -1,0,108,2,81,2,81,3,85,1,81,2,64,2,64,2,72,
2
500 DATA -1,0,53,2,72,2,72,3,85,1,85,2,96,2,108,4
510 DATA -1,0,53,2,53,2,53,2,81,2,61,2,64,2,64,2,72,2
520 DATA -1,0,53,2,53,2,53,2,81,2,61,2,64,2,64,2,72,2
530 DATA -1,0,47,2,47,2,47,2,53,2,61,2,64,2,61,4
540 DATA -1,0,72,2,64,1,61,1,53,3,81,1,81,2,72,2,64,4
550 DATA -1,0,47,3,47,1,47,2,53,2,61,2,64,2,61,4
560 DATA -1,0,72,2,64,1,61,1,53,3,81,1,81,2,72,2,81,6
570 DATA 0,0

Aantekening:

Aantekening:

Aantekening:

Aantekening: 7 00 00 00



In dit boek is een aantal zeer gevarieerde programma's opgenomen, waarin de kleur-, graphics- en geluidsmogelijkheden van de Atari 600XL/800XL optimaal worden benut.

- **Spelen in Arcada-stijl** om uw reactie en vingervlugheid te testen.
- **Avonturenspele**n om uw denkvermogen aan te scherpen en uw concentratie tot het uiterste te dwingen.
- **Puzzels en kansspelen**, voer voor de echte volhouders en goklustigen.
- **Vriendelijke toepassingen** om u te helpen uw uitgaven te regelen, uw vooruitgang in beeld te brengen en uw eigendommen te rubriceren.

Alle programma's in dit boek zijn geschikt voor de Atari 600XL en 800XL met 16k- of 64k-geheugen.